

AEROSPACE RECOMMENDED PRACTICE

SAE ARP4727

REV.
A

Issued 1995-10
Revised 2004-02

Superseding ARP4727

Gland Design, Computation of Seal Squeeze and Gland Volume

1. SCOPE:

- 1.1 This SAE Aerospace Recommended Practice (ARP) presents two BASIC language computer programs to promote and standardize the computation of installed O-ring cross-section deflection hereafter referred to as "squeeze" and the computation of gland volume. The two programs were written with line numbers and without use of any system specific BASIC commands to allow usage with as many systems as possible with a minimum of editing. The programs support entry of customary U.S. or metric dimensions.

The squeeze program, called SQ.BAS, has the following features:

- a. allows selection of either piston or rod gland
- b. allows entry of cap strip seal thickness
- c. computes minimum and maximum squeeze
- d. allows checking and correction of entries before execution
- e. allows viewing of output on screen before printing
- f. gives advisory messages for excessive stretch or inadequate squeeze
- g. allows on-line review of essential input and output variables prior to running program
- h. computes clearance between minimum gland height and maximum backup radial height when the backup dimension is entered
- i. program output is viewable on screen or may be printed out using either of two methods

SAE Technical Standards Board Rules provide that: "This report is published by SAE to advance the state of technical and engineering sciences. The use of this report is entirely voluntary, and its applicability and suitability for any particular use, including any patent infringement arising therefrom, is the sole responsibility of the user."

SAE reviews each technical report at least every five years at which time it may be reaffirmed, revised, or cancelled. SAE invites your written comments and suggestions.

Copyright © 2004 SAE International

All rights reserved. No part of this publication may be reproduced, stored in a retrieval system or transmitted, in any form or by any means, electronic, mechanical, photocopying, recording, or otherwise, without the prior written permission of SAE.

TO PLACE A DOCUMENT ORDER:

Tel: 877-606-7323 (inside USA and Canada)

Tel: 724-776-4970 (outside USA)

Fax: 724-776-0790

Email: custsvc@sae.org

SAE WEB ADDRESS:

<http://www.sae.org>

SAE ARP4727 Revision A

1.1 (Continued):

The gland volume program called GV.BAS, has the following features:

- a. allows selection of either piston or rod gland
- b. computes gland volume, O-ring volume and backup ring volume
- c. computes minimum and maximum percent occupancy
- d. allows viewing of output on screen before printing
- e. allows on-line review of essential input and output variables prior to running program
- f. allows entry of volume of a non-standard configuration or special seal such as a cap strip seal
- g. program output is viewable on screen or may be printed out using either of two methods

2. APPLICABLE DOCUMENTS:

The following publications form a part of this document to the extent specified herein. The latest issue of SAE publications shall apply. The applicable issue of other publications shall be the issue in effect on the date of the purchase order. In the event of conflict between the text of this document and references cited herein, the text of this document takes precedence. Nothing in this document, however, supersedes applicable laws and regulations unless a specific exemption has been obtained.

2.1 SAE Publications:

Available from SAE, 400 Commonwealth Drive, Warrendale, PA 15096-0001 Telephone: 724-776-4970, Web address: <http://www.sae.org>

AS568 Aerospace Size Standard for O-rings

AS4716 Aerospace Standard, Gland Design, O-ring and Other Elastomeric Seals

2.2 U.S. Government Publications:

Available from DODSSP Subscription Services Desk, Building 4D, 700 Robbins Avenue, Philadelphia, PA 19111-5094 Telephone: 215-697-2179, Web address: <http://assist.daps.mil/>

MIL-G-5514 Gland Design; Packing, Hydraulic, General Requirements For (inactive for new design)

SAE ARP4727 Revision A

3. TECHNICAL REQUIREMENTS:

3.1 Seal Squeeze:

- 3.1.1 General: This document gives a listing of a BASIC language computer program, SQ.BAS, which presents a standardized algorithm for computation of the deflection of the O-ring cross-section when installed in a gland. Work by several independent sources recognized the importance of maintaining squeeze between prescribed limits for maximum life of elastomeric seals. An upper bound is important to prevent deterioration, excessive friction, wear, and occasional spiral failure of O-rings. A lower bound is important to prevent premature compression set of the elastomer due to inadequate installed deflection, hereafter referred to as squeeze.
- 3.1.2 Background: MIL-G-5514 originated using inch fractional increments for gland dimensions and gradually evolved into a document based upon decimal increments which deviated somewhat from the original fractional increment organization. While the document was revised over the years to address various issues, no uniform application of squeeze requirements currently exists in the document. In 1985, the SAE Committee A-6 Seal Panel began work on a document to provide a standard which did provide a uniform standard for installed squeeze. The product was AS4716 which gives a squeeze not less than 0.005 inches when all adverse tolerances are considered, and also had a goal to not exceed 85 percent gland occupancy when minimum gland volume and maximum seal volumes are considered.. In some cases the occupancy exceeds 85 percent when all factors were considered.
- 3.1.3 Computational Basis: The equations used for program SQ.BAS are published in AS4716. The equations are based upon tests using O-rings, but are generally applicable to non-standard elastomeric seals as well, requiring only a knowledge of the seal inner diameter and radial cross-section dimensions in addition to the gland dimensions. AS4716 has tables for glands using AS568 O-ring dimensions which meet established criteria for squeeze and volume. Program SQ.BAS may be used for checking squeeze with other cross-sections and for salvage operations or in special circumstances, the gland and seal dimensions at extreme temperature conditions could be used to determine resultant squeeze at non-standard conditions. Appendix A gives a program listing for SQ.BAS.

SAE ARP4727 Revision A

3.1.4 Description of Operation:

3.1.4.1 Input of Data: When the program starts, a menu appears on the screen giving the following choices:

RUN (R)
QUIT (Q)
CHANGE DEFAULTS (C)
REVIEW VARIABLES (V)
ENTER CHOICE?

Entering "R" will begin main program.

Entering "Q" will end the program.

Entering "C" will give a menu to allow changing the default advisory values in the program. See 3.1.4.2 for further discussion of advisory default values.

Entering "V" will give a screen listing of the program variables.

Next, the entry of the following input data is requested:

0 GLAND (AS568 DASH NO. OR ISO O-RING CODE) - "; SIZE\$
1 TYPE - (R)OD OR (P)ISTON "; TYPE\$
2 SEAL CROSS SECTION - LOW VALUE "; WL
3 SEAL CROSS SECTION - HIGH VALUE "; WH
4 SEAL ID - LOW VALUE "; IDSL
5 SEAL ID - HIGH VALUE "; IDSH

If gland is piston type, the following two lines appear.

6 PISTON DIAMETER - LOW VALUE "; DPL
7 PISTON DIAMETER - HIGH VALUE "; DPH

If gland is rod type, the following two lines appear.

6 ROD DIAMETER - LOW VALUE "; DRL
7 ROD DIAMETER - HIGH VALUE "; DRH

The remaining lines are common to both applications.

8 BORE DIAMETER - LOW VALUE "; DBL
9 BORE DIAMETER - HIGH VALUE "; DBH
10 GROOVE DIAMETER - LOW VALUE "; DGL
11 GROOVE DIAMETER - HIGH VALUE "; DGH
12 GROOVE ECCENTRICITY "; ECC
13 CAP SEAL THICKNESS - LOW VALUE "; TCAPL
14 CAP SEAL THICKNESS - HIGH VALUE "; TCAPH
15 COMMENT: "; COMMENT\$

COMMENT\$ allows entry of a short note or comment on the case being run. It should be limited to 60 characters in order to fit on one line of the output format.

SAE ARP4727 Revision A

3.1.4.1 (Continued):

After the last carriage return, the screen shows a listing of the data entered and by choosing the number corresponding to the variable, a new value for any of the input data may be entered. Entering "A", when prompted, will allow entry of all data from the beginning, which sometimes is a time saver when many changes have to be made. Entering "N", when prompted, will execute computation portion of the program.

3.1.4.2 Output: The program output appears on the screen for review prior to printing. By choosing "R", the program will branch to the data review and correction menu. Choosing "P" causes the program output to be directed to a file. Entering "Q" stops the program.

There are two ways to obtain a hard copy of the program output.

1. Before running the program, create a new folder, SEALPRNT on Drive C:. Then, when the program prompts for printout, pressing "P" will direct the output to a editable text file called SQOUT.TXT which can be found at C:\SEALPRNT\.
SQOUT.TXT can be opened with word processor software for printing a hardcopy or cutting and pasting into other documents.
2. Another method of obtaining a hard copy of the output is to press the Print Scrn key while the output is shown on the screen, then open a new file on your word processor software and selecting PASTE from the Edit menu. The output will be pasted as editable text into the file.

Depending upon the default font used for the word processor, the printout will probably not be aligned as shown in the examples. To correct this, the printout format should be set to have a left margin of 1.0 inches and a right margin of 0.7 inches. Set the font to 10 point Courier New. These steps will align the columns and lines of text.

An example printout of the program for a -214 gland is shown on Figure 1. The columns entitled LOW and HIGH are the numerically low and high values entered for the seal and gland variables. The columns entitled AT MIN SQUEEZE and AT MAX SQUEEZE give the calculated results for the variables shown at the minimum squeeze and maximum squeeze conditions.

NOTE: The LOW and HIGH columns are not related to the variable usage for computed results in the AT MIN SQUEEZE and AT MAX SQUEEZE columns.

3.1.4.2 (Continued):

The program has several advisory messages which appear to highlight program results outside the normal accepted range of values for aerospace service gland parameters. Example advisory messages are:

CROSS-SECTION DEFLECTION IS INSUFFICIENT... MIN ALLOWED DEFL = 0.0050 inches

CROSS-SECTION SQUEEZE IS NOT EXCESSIVE... MAX ALLOWED SQUEEZE = 30.0 %

ID STRETCH IS OKAY... MAX ALLOWED STRETCH = 5.00 %

BACKUP HEIGHT IN GLAND IS OKAY... MAX BU = 0.1200 MIN GLAND = 0.1215

The default values are shown which are used for determining the text of the message to be printed. The messages and the default are advisory only, the program user is cautioned to research the industry accepted allowable values for any special circumstances when using the program. If different values are desired, select "C" at program opening menu to obtain a menu allowing the user to change the default minimum deflection, maximum percent squeeze, and maximum percent stretch. These values are advisory only and are used only in the messages at the end of the program output. Lines 480 thru 580 may be edited to change the text of the advisory messages. The program gives a message also when the stretch computed from the data entered exceeds the range established for the cross-section reduction equations. An example of the program output using metric gland dimensions is shown on Figure 2.

3.2 Gland Volume:

- 3.2.1 General: This standard gives a listing of a BASIC language computer program, GV.BAS, for the computation of gland volume, O-ring and backup ring volume and the minimum and maximum percent occupancy. The program algorithm uses exact volumes derived from volumes of revolution using integral calculus for the various shapes resulting from consideration of wall angles and corner radii.
- 3.2.2 Background: The importance of adequate gland volume has been recognized over the years to allow for seal expansion with temperature increase and for seal swell when subjected to hydraulic fluids. MIL-G-5514 made provisions for adequate volume. The work in AS4716 also considered adequate volume and uses a goal of not more than 85 percent occupancy with the most adverse tolerances and dimensions applied.

SAE ARP4727 Revision A

3.2.3 Computational Basis: Program GV.BAS computes the minimum and maximum volumes for the gland, O-ring, and backup rings. The volumes are combined to compute the minimum and maximum percent occupancy of the gland. The program allows entry of negative wall angle. Appendix B gives a listing for program GV.BAS. While the program is constructed to give the minimum and maximum percent occupancy for the dimensions entered, it may be used to compute the resultant volumes for non-standard seals and salvage situations by entering the same dimensions for the lower and upper values of each component. Appendix C shows the procedure used to derive the volumes for the geometric shapes which constitute a gland when broken down into the simplest geometric elements. The portions of the gland constituting volumes V1, V2, V3, and V4 in the output are shown in Appendix C for rod (ID) and piston (OD) applications. O-ring and backup ring dimension variables are shown in Appendix D.

3.2.4 Description of Operation:

3.2.4.1 Input of Data: When the program starts a message appears on the screen giving a choice of running the program or viewing a list of the program variables. Entering "V" will give a screen listing of the program variables. Entering "R" will begin main program. Next, the entry of the following input data is requested:

```
INPUT " 0 AS568 DASH NUMBER OR ISO O-RING SIZE CODE..... "; SIZE$
INPUT " 1 (R) OD OR (P)ISTON....."; TYPE$
INPUT " 2 GLAND WIDTH - LOW....."; WGL
INPUT " 3 GLAND WIDTH - HIGH....."; WGH
INPUT " 4 GROOVE DIAMETER - LOW....."; DGL
INPUT " 5 GROOVE DIAMETER - HIGH....."; DGH
INPUT " 6 GROOVE WALL ANGLE, DEGREES - LOW....."; AGL
INPUT " 7 GROOVE WALL ANGLE, DEGREES - HIGH....."; AGH
INPUT " 8 GROOVE CORNER RADIUS - LOW....."; RGL
INPUT " 9 GROOVE CORNER RADIUS - HIGH....."; RGH
```

IF TYPE\$ = "P" the following two lines appear

```
INPUT " 10 PISTON BORE DIAMETER - LOW....."; DBL
INPUT " 11 PISTON BORE DIAMETER - HIGH....."; DBH
```

IF TYPE\$ = "R" the following two lines appear

```
INPUT " 10 ROD DIAMETER - LOW....."; DRL
INPUT " 11 ROD DIAMETER - HIGH....."; DRH
```

```
INPUT " 12 O-RING CROSS SECTION DIAMETER - LOW....."; WSL
INPUT " 13 O-RING CROSS SECTION DIAMETER - HIGH....."; WSH
INPUT " 14 O-RING INNER DIAMETER - LOW....."; IDSL
INPUT " 15 O-RING INNER DIAMETER - HIGH....."; IDSH
INPUT " 16 NUMBER OF BACK UP RINGS....."; NB
```

SAE ARP4727 Revision A

3.2.4.1 (Continued):

IF NB is greater than zero, the following lines appear

```
INPUT " 17 BACK UP RING INNER DIAMETER - LOW....."; IDBUL
INPUT " 18 BACK UP RING INNER DIAMETER - HIGH....."; IDBUH
INPUT " 19 BACK UP RING RADIAL WIDTH - LOW....."; WBL
INPUT " 20 BACK UP RING RADIAL WIDTH - HIGH....."; WBH
INPUT " 21 BACK UP RING THICKNESS - LOW....."; TBL
INPUT " 22 BACK UP RING THICKNESS - HIGH....."; TBH
INPUT " 23 BACK UP RING GAP - LOW....."; GBUL
INPUT " 24 BACK UP RING GAP - HIGH....."; GBUH
INPUT " 25 BACK UP RING GAP ANGLE, DEGREES - LOW....."; ABUL
INPUT " 26 BACK UP RING GAP ANGLE, DEGREES - HIGH....."; ABUH
INPUT " 27 MIN VOLUME OF SPECIAL SEAL (ENTER 0 IF NONE).... "; VSPCL
INPUT " 28 MAX VOLUME OF SPECIAL SEAL (ENTER 0 IF NONE).... "; VSPCH
INPUT " 29 COMMENT: "; COMMENT$
```

NOTE: When entering VSPCL and VSPCH, be sure to use units consistent with dimension entries for other assembly components

3.2.4.2 Execution: When the program is run, a screen appears which gives a choice of "R" to run the program or "V" to view a listing of program variable definitions. If the "V" option is chosen, follow the prompts to complete viewing of the list. At the end of the list choose "R" to repeat list or "Q" to quit list and return to main program.

3.2.4.3 Program Output: When execution of the program is complete, an output screen appears which allows immediate viewing of results. An option appears giving a choice of "P" to print output to a file, "Q" to quit execution, or "R" to run again. If "R" is chosen a screen appears with the current values for variables to allow rerunning after changing only the values desired.

There are two ways to obtain a hard copy of the program output.

1. Before running the program, create a new folder, SEALPRNT on Drive C:. Then, when the program prompts for printout, pressing "P" will direct the output to a editable text file called GVOUT.TXT which can be found at C:\SEALPRNT\.
GVOUT.TXT can be opened with word processor software for printing a hardcopy or cutting and pasting into other documents.
2. Another method of obtaining hard copy of the output is to press the Print Scrn key while the output is shown on the screen, then open a new file on your word processor software and selecting PASTE from the Edit menu. The output will be pasted as editable text into the file. This operation will have to be performed twice, once for each screen of output.

SAE ARP4727 Revision A

3.2.4.3 (Continued):

Depending upon the default font used for the word processor, the printout will probably not be aligned as shown in the examples. To correct this, the printout format should be set to have a left margin of 1.0 inches and a right margin of 0.7 inches. Set the font to 10 point Courier New. These steps will align the columns and lines of text.

Figure 3 is an example of the output created using MIL-G-5514F gland dimensions. Figure 4 is an example output using metric gland dimensions. The columns entitled LOW and HIGH are the numerically low and high values entered for the seal and gland variables. The columns entitled MIN PCT and MAX PCT give the calculated results for the variables shown at the minimum squeeze and maximum occupancy conditions.

NOTE: The LOW and HIGH columns are not related to the variable usage for computed results in the MIN PCT and MAX PCT columns.

4. MEDIA AVAILABILITY:

The programs in this document are available on 3.5 inch floppy disks for personal computers from SAE.

5. NOTES:

5.1 Conversion of the BASIC listing to other Software:

For users who do not have access to a BASIC Language interpreter or compiler, the equations used for computing O-ring deflection and squeeze have been successfully used in spread sheet software by several individuals. Also, the equations can be entered into other available math software. The results should be checked against the examples in this document for accuracy. This would be a recommended solution for individuals lacking BASIC Language software or if personal preference are predominant considerations.

SAE ARP4727 Revision A

5.2 Changes to SQ.BAS for Revision A:

Changed or added lines in SQ.BAS are denoted by “ ‘REV A “ to the right of the line in the program listing.

A critical element of the program is the computation of reduction of O-ring cross-section due to stretch. ARP4727 originally used the quadratic equations from a published O-ring handbook. Beginning with revision ARP4727A, the equation used to compute the reduction in O-ring cross-section due to stretch uses a different equation, identified as the hyperbolic equation, for O-rings whose ID to cross-section ratio is equal to or less than 2.95. The original quadratic equations are still used for O-rings whose ID to cross-section ratio is greater than 2.95. The program results print a message to indicate the equation used, based upon the value of the ratio.

Another significant correction to the program is in the calculation of percent squeeze. The program has been corrected to compute the percent squeeze by taking the ratio of the installed deflection of the O-ring over the original free cross section diameter. This does not effect the computation of the O-ring deflection, but will change the calculated percent squeeze by a small amount from prior versions of the program.

A correction in the program was made regarding the allowance for cap seal thickness effect on stretch of a rod (ID) seal. The correct equation uses $2 \cdot TCAP$ on the diameter. The error, if unchanged, would only effect computation results if the squeeze calculation were being done using a cap seal in a rod (ID) seal installation.

Another revision was made in the definition of the variables STRL and STRH. These now refer to the low numerical value for stretch and the high numerical value for stretch respectively. This change does not effect results of the computed values.

Also, the program was changed so that when the default value for minimum squeeze in inches is altered, the corresponding value for ISO dimensions is revised to agree and vice versa. The default will remain as set until the program has been quit or restarted.

Improved the ease of getting printed output from the program.

5.3 Changes to GV.BAS for Revision A:

Changed or added lines in GV.BAS are denoted by “ ‘REV A “ to the right of the line in the program listing.

A minor correction was made (lines 4420-4440) to give the correct logic when the gland application case is changed from R to P or P to R.

Improved the ease of getting printed output from the program.

SAE ARP4727 Revision A

```

O-RING SEAL SQUEEZE - USING PGM SQ.BAS      VERSION 3.0

Time: 10:20:31 AM   Date: 03-20-2003

GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) -214   TYPE: PISTON (OD SEAL)
Example of -214 MIL-G-5514 piston seal squeeze.

IDSL/WL = 7.21  RATIO > 2.95:USED QUADRATIC EQUA FOR SEAL CROSS SECT REDUCT
IDSH/WH = 6.95  RATIO > 2.95:USED QUADRATIC EQUA FOR SEAL CROSS SECT REDUCT
*****

```

	LOW	HIGH
SEAL CROSS SECTION - in.	0.1350	0.1430
SEAL ID - in.	0.9740	0.9940
PISTON DIAMETER - in.	1.2380	1.2390
BORE DIAMETER - in.	1.2410	1.2430
GROOVE DIAMETER - in.	0.9960	0.9980
GROOVE ECCENTRICITY - in.	0.0000	0.0030

	AT MIN SQUEEZE	AT MAX SQUEEZE
SQUEEZE - PERCENT	3.66	16.71
O-RING DEFLECTION - in.	0.00494	0.02390
DIAMETRAL CLEARANCE - in.	0.0050	0.0030
STRETCH - PERCENT	2.26	0.40

```

***** ADVISORY MESSAGES *****
CROSS-SECT DEFLECTION IS INSUFFICIENT... MIN ALLOWED DEFL = 0.0050 inches
CROSS-SECTION SQUEEZE IS NOT EXCESSIVE... MAX ALLOWED SQUEEZE = 30.0 %
ID STRETCH IS OKAY... MAX ALLOWED STRETCH = 5.00 %
BACKUP HEIGHT IN GLAND IS OKAY... MAX BU = 0.1200 MIN GLAND = 0.1215
*****

```

FIGURE 1 - Example Printout from Program "SQ.BAS" Using SAE Dimensions

SAE ARP4727 Revision A

```

O-RING SEAL SQUEEZE - USING PGM SQ.BAS      VERSION 3.0

Time:  5:27:51 PM   Date: 09-26-2003

GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) -B0053A TYPE: PISTON (OD SEAL)
Example using metric gland dimensions.

IDSL/WL = 2.04  RATIO <= 2.95:USED HYPERBOLIC EQUA FOR SEAL CROSS SECT REDUCT
IDSH/WH = 1.98  RATIO <= 2.95:USED HYPERBOLIC EQUA FOR SEAL CROSS SECT REDUCT
*****

```

	LOW	HIGH
SEAL CROSS SECTION - mm	2.5400	2.7400
SEAL ID - mm	5.1700	5.4300
PISTON DIAMETER - mm	9.9200	9.9500
BORE DIAMETER - mm	10.0000	10.0700
GROOVE DIAMETER - mm	5.8300	5.8600
GROOVE ECCENTRICITY - mm	0.0000	0.0250

	AT MIN SQUEEZE	AT MAX SQUEEZE
SQUEEZE - PERCENT	7.26	22.71
O-RING DEFLECTION - mm	0.18438	0.62220
DIAMETRAL CLEARANCE - mm	0.1500	0.0800
STRETCH - PERCENT	12.77	7.92

```

***** ADVISORY MESSAGES *****
CROSS-SECTION DEFLECTION OKAY... MIN ALLOWED DEFL = 0.13 mm
CROSS-SECTION SQUEEZE IS NOT EXCESSIVE... MAX ALLOWED SQUEEZE = 30.0 %
NO ID STRETCH OR ID STRETCH EXCEEDS LIMIT... MAX ALLOWED STRETCH = 5.00 %
BACKUP HEIGHT TO GLAND CROSS-SECTION CLEARANCE NOT COMPUTED
*****

```

FIGURE 2 - Example Printout from Program "SQ.BAS" Using Metric Dimensions

SAE ARP4727 Revision A

```

GLAND VOLUME AND PERCENT OCCUPANCY - USING PGM GV.BAS

Time: 4:45:55 PM Date: 04-30-1995

GLAND IDENTIFICATION (AS568 OR ISO O-RING CODE) -214 TYPE: ROD (ID SEAL)
COMMENT: Example using AS4716-214 gland and MIL-R-8791/1-214 BACKUPS.

*****

```

	LOW	HIGH
GROOVE WIDTH-inches	0.3040	0.3140
GROOVE DIA-inches	1.2400	1.2420
GROOVE WALL ANGLE-DEG	0.0000	5.0000
GROOVE CORNER RADIUS-inches	0.0100	0.0250
ROD DIAMETER-inches	0.9960	0.9980
O-RING CROSS-SECTION DIA.-inches	0.1350	0.1430
O-RING INNER DIAMETER-inches	0.9740	0.9940
BACKUP RING INNER DIAMETER-inches	0.9970	0.9990
BACKUP RING WIDTH-inches	0.1180	0.1200
BACKUP RING THICKNESS-inches	0.0600	0.0640
BACKUP RING GAP-inches	0.0000	0.0060
BACKUP RING GAP ANGLE-DEG	19.0000	22.0000
NUMBER OF BACKUP RINGS	2	

	VOLUMES COMPUTED AND PERCENT OCCUPANCY	
	MIN PCT	MAX PCT
V1-in^3	0.191013E-02	0.000000E+00
V2-in^3	0.270170E-03	0.186785E-02
V3-in^3	0.127849E+00	0.108044E+00
V4-in^3	0.395532E-02	0.824857E-02
GLAND VOLUME-in^3	0.140120E+00	0.128276E+00
SEAL VOLUME-in^3	0.498699E-01	0.573683E-01
SPECIAL SEAL VOLUME-in^3	0.000000E+00	0.000000E+00
BACKUP VOLUME-in^3	0.246699E-01	0.269986E-01
PCT OCCUPANCY	70.80	86.82

```

*****

```

FIGURE 3 - Example Printout from Program "GV.BAS" Using SAE Dimensions

SAE ARP4727 Revision A

GLAND VOLUME AND PERCENT OCCUPANCY - USING PGM GV.BAS

Time: 5:27:49 PM Date: 04-30-1995

GLAND IDENTIFICATION (AS568 OR ISO O-RING CODE) -B0053A TYPE: PISTON (OD SEAL)
COMMENT: Example using metric gland and ISO3601-1 O-ring.

```
*****
                                LOW                HIGH
                                -----
GROOVE WIDTH-mm                3.6000            3.8500
GROOVE DIA-mm                  5.8100            5.8600
GROOVE WALL ANGLE-DEG          0.0000            5.0000
GROOVE CORNER RADIUS-mm        0.3000            0.5000
BORE DIAMETER-mm              10.0000           10.0700
O-RING CROSS-SECTION DIA.-mm   2.5400            2.7400
O-RING INNER DIAMETER-mm       5.1700            5.4300
NUMBER OF BACKUP RINGS         0
                                -----
                                VOLUMES COMPUTED AND PERCENT OCCUPANCY
                                MIN PCT            MAX PCT
                                -----
V1-mm^3                        0.418199E+01    0.000000E+00
V2-mm^3                        0.120978E+01    0.396979E+01
V3-mm^3                        0.175344E+03    0.134081E+03
V4-mm^3                        0.143144E+02    0.207896E+02
GLAND VOLUME-mm^3              0.214756E+03    0.183600E+03
SEAL VOLUME-mm^3               0.122733E+03    0.151343E+03
SPECIAL SEAL VOLUME-mm^3       0.000000E+00    0.000000E+00
BACKUP VOLUME-mm^3             0.000000E+00    0.000000E+00
PCT OCCUPANCY                  57.25            82.43
*****
```

FIGURE 4 - Example Printout from Program "GV.BAS" Using ISO Dimensions

PREPARED UNDER THE JURISDICTION OF
SAE COMMITTEE A6C, SEALS PANEL OF
COMMITTEE A-6, AEROSPACE FLUID POWER, ACTUATION & CONTROL TECHNOLOGIES

SAE ARP4727 Revision A

APPENDIX A LISTING OF "SQ.BAS"

Line revisions are denoted by 'Rev A within program

```

10 REM * * * * *
20 REM *      PROGRAM SQ.BAS - O-RING SEAL SQUEEZE      VERSION 3.0      *
30 REM * THIS IS A PROGRAM TO CALCULATE THE MINIMUM AND MAXIMUM INSTALLED *
40 REM * SQUEEZE AND STRETCH OF AN O-RING OR OTHER SHAPE ELASTOMERIC SEAL. *
50 REM *
60 REM *      VERSION 3.0 HAS THE FOLLOWING CAPABILITIES:
70 REM * (A) COMPUTES SQUEEZE OF AN O-RING ENERGIZER FOR A CAP STRIP SEAL *
80 REM * (B) CHECKS FOR ACCEPTABLE MINIMUM AND MAXIMUM SQUEEZE BASED UPON *
90 REM *      DEFAULT VALUES WITHIN THE PROGRAM
100 REM* (C) CHECKS FOR EXCESSIVE STRETCH
110 REM* (D) CHECKS FOR INTERFERENCE OF MINIMUM GLAND WITH MAXIMUM BACKUP *
120 REM* RING HEIGHT IF BACKUP HEIGHT ENTERED
130 REM* (E) COMBINES ROD (ID SEAL) OR PISTON (OD SEAL) TYPE
140 REM* INTO ONE PROGRAM - REQUIRES CHECK AND CORRECTION
150 REM* OF CHANGED VALUES.
160 REM* (F) SUPPORTS SAE OR ISO GLAND DIMENSIONS
170 REM* (G) HARD COPY OF OUTPUT MAY BE OBTAINED BY METHODS 1 OR 2 BELOW. * 'REV A
180 REM* (1) BEFORE RUNNING SQ.BAS, CREATE A FOLDER CALLED SEALPRNT * 'REV A
190 REM* ON DRIVE C:\. THE PROGRAM WILL PRINT TO A FILE CALLED * 'REV A
200 REM* SQOUT.TXT WHEN THE OPTION "P" IS PROMPTED. * 'REV A
210 REM* (2) A HARD COPY MAY ALSO BE OBTAINED BY PRESSING THE * 'REV A
220 REM* PRINT SCR N KEY ON THE COMPUTER KEYBOARD WHILE THE SCREEN * 'REV A
230 REM* IS DISPLAYED. THEN OPEN A NEW FILE IN WORD PROCESSOR * 'REV A
240 REM* SOFTWARE, PASTE THE OUTPUT TO THE WORD PROCESSOR FILE. * 'REV A
250 REM* EITHER METHOD ABOVE WILL GIVE AN EDITABLE TEXT FILE WHICH * 'REV A
260 REM* CAN BE PRINTED FROM THE WORD PROCESSOR FILE OR CUT * 'REV A
270 REM* AND PASTED INTO ANOTHER FILE IF DESIRED. * 'REV A
280 REM* (H) THIS VERSION USES THE HYPERBOLIC EQUATION TO COMPUTE THE * 'REV A
290 REM* REDUCTION IN SEAL CROSS-SECTION DUE TO STRETCH FOR SEAL * 'REV A
300 REM* ID TO SEAL CROSS-SECTION RATIOS OF 2.95 OR LESS. * 'REV A
310 REM* (I) REV 3: PERCENT SQUEEZE CHANGED TO RATIO OF DEFLECTION OVER * 'REV A
320 REM* FREE O-RING CROSS SECT DIA. REVERSED DEFINITION OF STRL AND * 'REV A
330 REM* STRH TO REFLECT MIN STRETCH AND MAX STRETCH NUMERICALLY. * 'REV A
340 REM* CORRECTED ROD SEAL STRETCH TO USE 2*TCAPL AND 2*TCAPH * 'REV A
350 REM* ORIGINALLY WRITTEN FEBRUARY 1993 BY R.B. OLSEN
360 REM* FOR ARP4727, SAE COMMITTEE A6 PROJECT A6C2-91-2
370 REM* REVISED: JANUARY 2003, A6 PROJECT A6C2-01-03
380 REM* * * * *
390 REM=====
400 REM      CLEAR SCREEN
410 REM=====
420 CLS
430 REM=====
440 REM      INITIALIZE VARIABLES
450 REM=====
460 DEFLOI = .005: DEFLOM = .127: SQUZHI = 30: STRCHI = 5: ISOWL = 1.7 'REV A
470 VER$ = "3.0": Q1 = 0: Q2 = 0: Q3 = 0: Q4 = 0: Q5 = 0: Q6 = 0 'REV A
480 MSG1$(0) = "CROSS-SECTION DEFLECTION OKAY..."
490 MSG2$(0) = "CROSS-SECTION SQUEEZE IS NOT EXCESSIVE..."
500 MSG3$(0) = "ID STRETCH IS OKAY..."
510 MSG4$(0) = "BACKUP HEIGHT IN GLAND IS OKAY..."
520 MSG1$(1) = "CROSS-SECT DEFLECTION IS INSUFFICIENT..."
530 MSG2$(1) = "CROSS-SECTION SQUEEZE EXCEEDS LIMIT..."
540 MSG3$(1) = "NO ID STRETCH OR ID STRETCH EXCEEDS LIMIT..."
550 MSG4$(1) = "BACKUP HEIGHT EXCEEDS GLAND..."
560 MSG4$(2) = "BACKUP HEIGHT TO GLAND CROSS-SECTION CLEARANCE NOT COMPUTED"
570 MSG5$(1) = "RATIO <= 2.95:USED HYPERBOLIC EQUA FOR SEAL CROSS SECT REDUCT" 'REV A
580 MSG5$(2) = "RATIO > 2.95:USED QUADRATIC EQUA FOR SEAL CROSS SECT REDUCT" 'REV A
590 REM=====
600 REM      PRINT SCREEN HEADING AND
610 REM      INPUT VARIABLES
620 REM=====
630 CLS : LOCATE 1: PRINT TAB(6); : PRINT "PROGRAM SQ.BAS - O-RING SEAL SQUEEZE"; : PRINT
TAB(51); : PRINT "VERSION "; VER$;
640 LOCATE 4: PRINT TAB(5); "FOR PROGRAM DESCRIPTION AND VARIABLE LISTING SEE ARP4727"

```

SAE ARP4727 Revision A

```

650 LOCATE 6: PRINT TAB(5); "AVAILABLE FROM: SOCIETY OF AUTOMOTIVE ENGINEERS, WARRENDALE, PA"
660 LOCATE 11: PRINT TAB(10); "RUN (R)"
670 LOCATE 12: PRINT TAB(10); "QUIT (Q)"
680 LOCATE 13: PRINT TAB(10); "CHANGE DEFAULTS (C)"
690 LOCATE 14: PRINT TAB(10); "REVIEW VARIABLES (V)"
700 LOCATE 15: PRINT TAB(10); "ENTER CHOICE"; : INPUT A$
710 IF A$ = "R" OR A$ = "r" THEN GOTO 750
720 IF A$ = "Q" OR A$ = "q" THEN END
730 IF A$ = "C" OR A$ = "c" THEN GOTO 4740
740 GOTO 4200
750 CLS : LOCATE 1
760 Q1 = 0: Q2 = 0: Q3 = 0: Q4 = 0: Q5 = 0: Q6 = 0'REV A
770 PRINT TAB(6); : PRINT "PROGRAM SQ.BAS - O-RING SEAL SQUEEZE"; : PRINT TAB(51); : PRINT
"VERSION "; VER$; : PRINT
780 INPUT " 0 GLAND (AS568 DASH NO. OR ISO O-RING CODE) -"; SIZE$
790 INPUT " 1 TYPE - (R)OD (ID SEAL) OR (P)ISTON (OD SEAL) "; TYPE$
800 IF TYPE$ = "r" THEN TYPE$ = "R"
810 IF TYPE$ = "p" THEN TYPE$ = "P"
820 INPUT " 2 SEAL CROSS SECTION - LOW VALUE "; WL
830 INPUT " 3 SEAL CROSS SECTION - HIGH VALUE "; WH
840 INPUT " 4 SEAL ID - LOW VALUE "; IDSL
850 INPUT " 5 SEAL ID - HIGH VALUE "; IDSH
860 IF TYPE$ = "R" GOTO 900
870 INPUT " 6 PISTON DIAMETER - LOW VALUE "; DPL
880 INPUT " 7 PISTON DIAMETER - HIGH VALUE "; DPH
890 GOTO 920
900 INPUT " 6 ROD DIAMETER - LOW VALUE "; DRL
910 INPUT " 7 ROD DIAMETER - HIGH VALUE "; DRH
920 INPUT " 8 BORE DIAMETER - LOW VALUE "; DBL
930 INPUT " 9 BORE DIAMETER - HIGH VALUE "; DBH
940 INPUT " 10 GROOVE DIAMETER - LOW VALUE "; DGL
950 INPUT " 11 GROOVE DIAMETER - HIGH VALUE "; DGH
960 INPUT " 12 GROOVE ECCENTRICITY "; ECC
970 INPUT " 13 CAP SEAL THICKNESS - LOW VALUE "; TCAPL
980 INPUT " 14 CAP SEAL THICKNESS - HIGH VALUE "; TCAPH
990 INPUT " 15 BACKUP RING MAX RADIAL HEIGHT 'W' (OR ENTER 0) "; WBUH
1000 INPUT " 16 COMMENT:"; COMMENT$
1010 REM=====
1020 REM CALL SUBR TO CHECK INPUT VALUES
1030 REM=====
1040 GOSUB 3360
1050 Q1 = 0: Q2 = 0: Q3 = 0: Q4 = 0: Q5 = 0: Q6 = 0'REV A
1060 IF L$ = "A" OR L$ = "a" THEN CLS : GOTO 470
1070 REM=====
1080 REM COMPUTE MIN AND MAX STRETCH
1090 REM=====
1100 IF TYPE$ = "R" GOTO 1140
1110 STRH = (DGL - IDSL) / IDSL 'REV A
1120 STRL = (DGH - IDSH) / IDSH 'REV A
1130 GOTO 1160
1140 STRH = (DRL + 2 * TCAPL - IDSL) / IDSL 'REV A
1150 STRL = (DRH + 2 * TCAPH - IDSH) / IDSH 'REV A
1160 REM=====
1170 REM COMPUTE GLAND CROSS SECTIONS
1180 REM=====
1190 IF TYPE$ = "R" GOTO 1230
1200 GLNDMX = (DBH - DGL) / 2
1210 GLNDMN = (DBL - DGH) / 2
1220 GOTO 1250
1230 GLNDMX = (DGH - DRL) / 2
1240 GLNDMN = (DGL - DRH) / 2
1250 GLNDH = GLNDMX - TCAPL
1260 GLNDL = GLNDMN - TCAPH
1270 REM=====
1280 REM COMPUTE DIAMETRAL AND
1290 REM RADIAL CLEARANCE
1300 REM=====
1310 IF TYPE$ = "R" GOTO 1350
1320 DCLRMX = DBH - DPL
1330 DCLRMN = DBL - DPL
1340 GOTO 1370

```

SAE ARP4727 Revision A

```

1350 DCLRMX = DBH - DRL
1360 DCLRMN = DBH - DRH
1370 RCLRMX = DCLRMX / 2
1380 RCLRMN = DCLRMN / 2
1390 REM=====
1400 REM  COMPUTE RADIAL ECCENTRICITY
1410 REM=====
1420 RECC = ECC / 2
1430 REM=====
1440 REM  DEFINE CROSS-SECTION REDUCTION
1450 REM  FUNCTIONS FOR STRETCH <= .03 AND STRETCH > .03
1460 REM      AND SEAL ID/SEAL XSECT > 2.95  'REV A
1470 REM=====
1480 DEF FNLE03 (X) = .0001 + 1.06 * X - 10 * X ^ 2
1490 DEF FNGT03 (X) = .0056 + .59 * X - .46 * X ^ 2
1500 REM=====
1510 REM TEST FOR STRETCH OUT OF RANGE
1520 REM=====
1530 IF STRH > 25 / 100 OR STRL > 25 / 100 THEN 3150  'REV A
1540 REM=====
1550 REM  COMPUTE CROSS SECTION
1560 REM  REDUCTION FACTOR FOR STRH
1570 REM  IF RATIO <= 2.95 HYPERBOLIC  'REV A
1580 REM  EQUATION IS USED  'REV A
1590 REM=====
1600 IF STRH < 0 THEN DELWL = 0: GOTO 1690  'REV A
1610 RATIO1 = IDSL / WL
1620 IF RATIO1 <= 2.95 THEN DELWL = STRH * 100 / (315 * EXP(-(RATIO1)) + 165 + STRH * 100): GOTO
1690  'REV A
1630 IF STRH > .03 THEN DELWL = FNGT03(STRH): GOTO 1690  'REV A
1640 DELWL = FNLE03(STRH)  'REV A
1650 REM=====
1660 REM  COMPUTE CROSS SECTION
1670 REM  REDUCTION FACTOR FOR STRL
1680 REM=====
1690 IF STRL < 0 THEN DELWH = 0: GOTO 1770  'REV A
1700 RATIO2 = IDSH / WH
1710 IF RATIO2 <= 2.95 THEN DELWH = STRL * 100 / (315 * EXP(-(RATIO2)) + 165 + STRL * 100): GOTO
1770  'REV A
1720 IF STRL > .03 THEN DELWH = FNGT03(STRL): GOTO 1770  'REV A
1730 DELWH = FNLE03(STRL)  'REV A
1740 REM=====
1750 REM  COMPUTE CROSS SECTION  'REV A
1760 REM  DEFLECTION  'REV A
1770 REM=====
1780 WMIN = WL - DELWL * WL: WMAX = WH - DELWH * WH
1790 DEFMN = WL - DELWL * WL - (GLNDH + RCLRMX + RECC)
1800 DEFMX = WH - DELWH * WH - (GLNDL - RCLRMN - RECC)
1810 REM=====
1820 REM  COMPUTE SQUEEZE  'REV A
1830 REM=====
1840 SQEL = DEFMN * 100 / WL  'REV A
1850 SQEH = DEFMX * 100 / WH  'REV A
1860 REM=====
1870 REM  TEST FOR MIN AND MAX SQUEEZE
1880 REM=====
1890 IF WL >= ISOWL THEN DEFLO = DEFLOM ELSE DEFLO = DEFLOI
1900 IF DEFMN < DEFLO THEN Q1 = 1
1910 IF SQEH > SQUZHI THEN Q2 = 1
1920 REM=====
1930 REM  TEST FOR EXCESSIVE OR NO STRETCH
1940 REM=====
1950 IF STRL < 0 THEN Q3 = 1
1960 IF STRH < 0 THEN Q3 = 1
1970 IF STRL > STRCHI / 100 THEN Q3 = 1
1980 IF STRH > STRCHI / 100 THEN Q3 = 1
1990 REM=====
2000 REM TEST FOR GLAND INTERFERENCE
2010 REM  WITH BACKUP HEIGHT
2020 REM=====
2030 IF WBUH = 0 THEN Q4 = 2: GOTO 2090

```

SAE ARP4727 Revision A

```

2040 BUCLR = INT((GLNDMN * 10 ^ 6 + .5) / 10 ^ 6 - WBUH)'FORCE TRUNCATION OF ROUND OFF
2050 IF BUCLR < 0 THEN Q4 = 1
2060 REM=====
2070 REM      PRINT RESULTS TO SCREEN
2080 REM=====
2090 IF WL > ISOWL THEN DM$ = " - mm" ELSE DM$ = " - in."
2100 CLS : LOCATE 1
2110 TYPE1$ = TYPE$ + "OD (ID SEAL)": IF TYPE$ = "P" THEN TYPE1$ = TYPE$ + "ISTON (OD SEAL)"
2120 PRINT TAB(8); : PRINT "O-RING SEAL SQUEEZE - USING PGM SQ.BAS"; : PRINT TAB(56); : PRINT
"VERSION "; VER$;
2130 PRINT TAB(5); "GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) -"; SIZE$; TAB(58); "TYPE: ";
TYPE1$
2140 PRINT TAB(5); "COMMENT: "; COMMENT$
2150 PRINT TAB(5); "IDSL/WL="; : PRINT USING "##.##"; IDSL / WL; : PRINT TAB(20); "IDSH/WH="; :
PRINT USING "##.##"; IDSH / WH; : PRINT TAB(42); "  LOW"; TAB(62); "  HIGH"
2160 PRINT TAB(5); "IF RATIO <= 2.95 USES HYPBLC EQUA"; : PRINT TAB(40); "-----"; TAB(60); "-
-----" 'REV A
2170 PRINT TAB(5); "SEAL CROSS SECTION"; DM$; TAB(40); : PRINT USING "####.####"; WL; : PRINT
TAB(60); : PRINT USING "####.####"; WH
2180 PRINT TAB(5); "SEAL ID"; DM$; TAB(40); : PRINT USING "####.####"; IDSL; : PRINT TAB(60); :
PRINT USING "####.####"; IDSH
2190 IF TYPE$ = "R" GOTO 2220
2200 PRINT TAB(5); "PISTON DIAMETER"; DM$; TAB(40); : PRINT USING "####.####"; DPL; : PRINT
TAB(60); : PRINT USING "####.####"; DPH
2210 GOTO 2230
2220 PRINT TAB(5); "ROD DIAMETER"; DM$; TAB(40); : PRINT USING "####.####"; DRL; : PRINT TAB(60);
: PRINT USING "####.####"; DRH
2230 PRINT TAB(5); "BORE DIAMETER"; DM$; TAB(40); : PRINT USING "####.####"; DBL; : PRINT
TAB(60); : PRINT USING "####.####"; DBH
2240 PRINT TAB(5); "GROOVE DIAMETER"; DM$; TAB(40); : PRINT USING "####.####"; DGL; : PRINT
TAB(60); : PRINT USING "####.####"; DGH
2250 PRINT TAB(5); "GROOVE ECCENTRICITY"; DM$; TAB(40); "  0.0000"; TAB(60); : PRINT USING
"####.####"; ECC
2260 IF TCAPL + TCAPH = 0 GOTO 2280
2270 PRINT TAB(5); "CAP SEAL THICKNESS"; DM$; TAB(40); : PRINT USING "####.####"; TCAPL; : PRINT
TAB(60); : PRINT USING "####.####"; TCAPH
2280 PRINT TAB(41); "AT MIN SQZ"; TAB(61); "AT MAX SQZ"
2290 PRINT TAB(40); "-----"; TAB(60); "-----"
2300 PRINT TAB(5); "SQUEEZE PERCENT"; DM$; TAB(40); : PRINT USING "####.##"; SQEL; : PRINT TAB(60); :
PRINT USING "####.##"; SQEH
2310 PRINT TAB(5); "O-RING DEFLECTION"; DM$; TAB(40); : PRINT USING "####.####"; DEFMN; : PRINT
TAB(60); : PRINT USING "####.####"; DEFMX
2320 PRINT TAB(5); "DIAMETRAL CLEARANCE"; DM$; TAB(40); : PRINT USING "####.####"; DCLRMX; :
PRINT TAB(60); : PRINT USING "####.####"; DCLRMN
2330 PRINT TAB(5); "STRETCH PERCENT"; TAB(40); : PRINT USING "####.##"; STRH * 100; : PRINT
TAB(60); : PRINT USING "####.##"; STRL * 100 'REV A
2340 PRINT TAB(25); "***** ADVISORY MESSAGES *****"
2350 IF WL >= ISOWL THEN GOTO 2380
2360 PRINT MSG1$(Q1); "  MIN ALLOWED DEFL = "; : PRINT USING "##.###"; DEFLO; : PRINT " inches"
2370 GOTO 2390
2380 PRINT MSG1$(Q1); "  MIN ALLOWED DEFL = "; : PRINT USING "##.##"; DEFLO; : PRINT " mm"
2390 PRINT MSG2$(Q2); "  MAX ALLOWED SQUEEZE = "; : PRINT USING "####.##"; SQUZHI; : PRINT " %"
2400 PRINT MSG3$(Q3); "  MAX ALLOWED STRETCH = "; : PRINT USING "####.##"; STRCHI; : PRINT " %"
2410 IF Q4 = 2 THEN GOTO 2440
2420 PRINT MSG4$(Q4); "  MAX BU = "; : PRINT USING "####.####"; WBUH; : PRINT "  MIN GLAND = "; :
PRINT USING "####.####"; GLNDMN
2430 GOTO 2450
2440 PRINT MSG4$(Q4)
2450 PRINT TAB(25); "***** ANALYSIS COMPLETE *****"
2460 REM=====
2470 REM      OPTIONS MENU
2480 REM=====
2490 LOCATE 24
2500 PRINT TAB(10); : PRINT "CHOOSE OPTION DESIRED ";
2510 INPUT "(R)ERUN (P)RINT (Q)UIT "; B$
2520 IF B$ = "R" OR B$ = "r" THEN 1040
2530 IF B$ = "P" OR B$ = "p" THEN 2570
2540 IF B$ = "Q" OR B$ = "q" THEN 3310
2550 IF ((B$ <> "R") AND (B$ <> "P") AND (B$ <> "Q")) THEN 2100
2560 REM=====
2570 REM      SEND RESULTS TO PRINTER

```

SAE ARP4727 Revision A

```

2580 REM=====
2590 OPEN "C:\SEALPRNT\SQOUT.TXT" FOR OUTPUT AS #1
2600 T$ = TIME$
2610 D$ = DATE$
2620 HR = VAL(T$)
2630 IF HR < 12 THEN AMPM$ = " AM" ELSE AMPM$ = " PM"
2640 IF HR > 12 THEN HR = HR - 12
2650 PRINT #1, TAB(13); "O-RING SEAL SQUEEZE - USING PGM SQ.BAS"; TAB(56); "VERSION "; VER$
2660 PRINT #1,
2670 PRINT #1, TAB(8); "Time: "; STR$(HR); RIGHT$(T$, 6); AMPM$; TAB(28); "Date: "; D$
2680 PRINT #1,
2690 PRINT #1, TAB(5); "GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) -"; SIZE$; TAB(58); "TYPE:
"; TYPE1$
2700 PRINT #1, TAB(5); COMMENT$
2710 PRINT #1,
2720 IF RATIO1 <= 2.95 THEN Q5 = 1 ELSE Q5 = 2
2730 IF RATIO2 <= 2.95 THEN Q6 = 1 ELSE Q6 = 2:
2740 PRINT #1, TAB(5); "IDSL/WL ="; : PRINT #1, USING "###.##"; RATIO1; : PRINT #1, TAB(21);
MSG5$(Q5) 'REV A
2750 PRINT #1, TAB(5); "IDSH/WH ="; : PRINT #1, USING "###.##"; RATIO2; : PRINT #1, TAB(21);
MSG5$(Q6) 'REV A
2760 PRINT #1, : FOR I = 0 TO 79: PRINT #1, "*"; : NEXT I
2770 PRINT #1,
2780 PRINT #1, TAB(40); " LOW"; TAB(60); " HIGH"
2790 PRINT #1, TAB(40); "-----"; TAB(60); "-----"
2800 PRINT #1, TAB(5); "SEAL CROSS SECTION"; DM$; TAB(40); : PRINT #1, USING "####.####"; WL; :
PRINT #1, TAB(60); : PRINT #1, USING "####.####"; WH
2810 PRINT #1, TAB(5); "SEAL ID"; DM$; TAB(40); : PRINT #1, USING "####.####"; IDSL; : PRINT #1,
TAB(60); : PRINT #1, USING "####.####"; IDSH
2820 IF TYPE$ = "R" GOTO 2850
2830 PRINT #1, TAB(5); "PISTON DIAMETER"; DM$; TAB(40); : PRINT #1, USING "####.####"; DPL; :
PRINT #1, TAB(60); : PRINT #1, USING "####.####"; DPH
2840 GOTO 2860
2850 PRINT #1, TAB(5); "ROD DIAMETER"; DM$; TAB(40); : PRINT #1, USING "####.####"; DRL; : PRINT
#1, TAB(60); : PRINT #1, USING "####.####"; DRH
2860 PRINT #1, TAB(5); "BORE DIAMETER"; DM$; TAB(40); : PRINT #1, USING "####.####"; DBL; : PRINT
#1, TAB(60); : PRINT #1, USING "####.####"; DBH
2870 PRINT #1, TAB(5); "GROOVE DIAMETER"; DM$; TAB(40); : PRINT #1, USING "####.####"; DGL; :
PRINT #1, TAB(60); : PRINT #1, USING "####.####"; DGH
2880 PRINT #1, TAB(5); "GROOVE ECCENTRICITY"; DM$; TAB(40); " 0.0000"; : PRINT #1, TAB(60); :
PRINT #1, USING "####.####"; ECC
2890 IF TCAPL + TCAPH = 0 GOTO 2910
2900 PRINT #1, TAB(5); "CAP SEAL THICKNESS"; DM$; TAB(40); : PRINT #1, USING "####.####"; TCAPL;
: PRINT #1, TAB(60); : PRINT #1, USING "####.####"; TCAPH
2910 PRINT #1,
2920 PRINT #1, TAB(42); "AT MIN"; TAB(62); "AT MAX"
2930 PRINT #1, TAB(42); "SQUEEZE"; TAB(62); "SQUEEZE"
2940 PRINT #1, TAB(40); "-----"; TAB(60); "-----"
2950 PRINT #1, TAB(5); "SQUEEZE - PERCENT"; TAB(40); : PRINT #1, USING "####.##"; SQEL; : PRINT
#1, TAB(60); : PRINT #1, USING "####.##"; SQEH
2960 PRINT #1, TAB(5); "O-RING DEFLECTION"; DM$; TAB(40); : PRINT #1, USING "####.####"; DEFMN;
: PRINT #1, TAB(60); : PRINT #1, USING "####.####"; DEFMX
2970 PRINT #1, TAB(5); "DIAMETRAL CLEARANCE"; DM$; TAB(40); : PRINT #1, USING "####.####";
DCLRMX; : PRINT #1, TAB(60); : PRINT #1, USING "####.####"; DCLRMN
2980 PRINT #1, TAB(5); "STRETCH - PERCENT"; TAB(40); : PRINT #1, USING "####.##"; STRH * 100; :
PRINT #1, TAB(60); : PRINT #1, USING "####.##"; STRL * 100 'REV A
2990 PRINT #1,
3000 PRINT #1, TAB(25); "***** ADVISORY MESSAGES *****"
3010 IF WL >= ISOWL THEN GOTO 3040
3020 PRINT #1, TAB(3); MSG1$(Q1); " MIN ALLOWED DEFL = "; : PRINT #1, USING "#.####"; DEFLO; :
PRINT #1, " inches"
3030 GOTO 3050
3040 PRINT #1, TAB(3); MSG1$(Q1); " MIN ALLOWED DEFL = "; : PRINT #1, USING "#.##"; DEFLO; :
PRINT #1, " mm"
3050 PRINT #1, TAB(3); MSG2$(Q2); " MAX ALLOWED SQUEEZE = "; : PRINT #1, USING "####.##"; SQUZHI;
: PRINT #1, " %"
3060 PRINT #1, TAB(3); MSG3$(Q3); " MAX ALLOWED STRETCH = "; : PRINT #1, USING "####.##"; STRCHI;
: PRINT #1, " %"
3070 IF Q4 = 2 THEN GOTO 3100
3080 PRINT #1, TAB(3); MSG4$(Q4); " MAX BU = "; : PRINT #1, USING "####.####"; WBUH; : PRINT #1, "
MIN GLAND = "; : PRINT #1, USING "####.####"; GLNDMN

```

SAE ARP4727 Revision A

```

3090 GOTO 3110
3100 PRINT #1, TAB(3); MSG4$(Q4)
3110 PRINT #1, : FOR I = 0 TO 79: PRINT #1, " "; : NEXT I
3120 PRINT #1,: CLOSE 'REV A
3130 GOTO 3280
3140 REM=====
3150 REM  ADVISORY MESSAGE FOLLOWS
3160 REM=====
3170 CLS : LOCATE 1
3180 PRINT
3190 PRINT TAB(15); "***** STRETCH OUT OF RANGE OF THIS PROGRAM *****"
3200 PRINT
3210 PRINT TAB(5); "      STRETCH PERCENT - LOW VALUE      "; STRL * 100
3220 PRINT TAB(5); "      STRETCH PERCENT - HIGH VALUE     "; STRH * 100
3230 PRINT : PRINT
3240 REM=====
3250 REM  CHANCE TO RERUN WITHOUT
3260 REM  RE-ENTERING ALL VARIABLES
3270 REM=====
3280 LOCATE 23: PRINT TAB(10); : INPUT "DO YOU WANT TO RUN AGAIN WITH VARIATIONS? (Y OR N) "; B$
3290 IF B$ = "N" OR B$ = "n" THEN 3310
3300 IF B$ = "Y" OR B$ = "y" THEN 1040
3310 END
3320 REM=====
3330 REM  SCREEN LISTING OF INPUTS
3340 REM  AND CORRECTION SUBROUTINE
3350 REM=====
3360 CLS : LOCATE 1
3370 PRINT TAB(20); : PRINT "PROGRAM SQ.BAS - O-RING SEAL SQUEEZE"; : PRINT
3380 PRINT TAB(5); "0  GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) . . . . - "; SIZE$
3390 PRINT TAB(5); "1  TYPE APPLICATION . . . . ."; TYPE$
3400 PRINT TAB(5); "2  SEAL CROSS SECTION - LOW VALUE . . . . ."; : PRINT USING
"#####.#####"; WL
3410 PRINT TAB(5); "3  SEAL CROSS SECTION - HIGH VALUE. . . . ."; : PRINT USING
"#####.#####"; WH
3420 PRINT TAB(5); "4  SEAL ID - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; IDSL
3430 PRINT TAB(5); "5  SEAL ID - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; IDSH
3440 IF TYPE$ = "R" GOTO 3480
3450 PRINT TAB(5); "6  PISTON DIAMETER - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DPL
3460 PRINT TAB(5); "7  PISTON DIAMETER - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DPH
3470 GOTO 3500
3480 PRINT TAB(5); "6  ROD DIAMETER - LOW VALUE . . . . ."; : PRINT USING
"#####.#####"; DRL
3490 PRINT TAB(5); "7  ROD DIAMETER - HIGH VALUE. . . . ."; : PRINT USING
"#####.#####"; DRH
3500 PRINT TAB(5); "8  BORE DIAMETER - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DBL
3510 PRINT TAB(5); "9  BORE DIAMETER - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DBH
3520 PRINT TAB(4); "10 GROOVE DIAMETER - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DGL
3530 PRINT TAB(4); "11 GROOVE DIAMETER - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DGH
3540 PRINT TAB(4); "12 GROOVE ECCENTRICITY. . . . ."; : PRINT USING
"#####.#####"; ECC
3550 PRINT TAB(4); "13 CAP SEAL THICKNESS - LOW VALUE . . . . ."; : PRINT USING
"#####.#####"; TCAPL
3560 PRINT TAB(4); "14 CAP SEAL THICKNESS - HIGH VALUE. . . . ."; : PRINT USING
"#####.#####"; TCAPH
3570 PRINT TAB(4); "15 BACKUP RING RADIAL HEIGHT - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; WBUH
3580 PRINT TAB(4); "16 COMMENT: "; COMMENT$
3590 LOCATE 20: PRINT "      DO YOU WANT TO CHANGE SOME VALUES?"
3600 INPUT "ENTER 0 TO 16 FOR ITEM, 'A' FOR ALL OR 'N' FOR NO CHANGE AND EXECUTE"; L$
3610 IF L$ = "N" OR L$ = "n" THEN RETURN
3620 IF L$ = "0" GOTO 3810
3630 IF L$ = "1" GOTO 3820

```

SAE ARP4727 Revision A

```

3640 IF L$ = "2" GOTO 4000
3650 IF L$ = "3" GOTO 4010
3660 IF L$ = "4" GOTO 4020
3670 IF L$ = "5" GOTO 4030
3680 IF L$ = "6" GOTO 4040
3690 IF L$ = "7" GOTO 4060
3700 IF L$ = "8" GOTO 4100
3710 IF L$ = "9" GOTO 4110
3720 IF L$ = "10" GOTO 4120
3730 IF L$ = "11" GOTO 4130
3740 IF L$ = "12" GOTO 4140
3750 IF L$ = "13" GOTO 4150
3760 IF L$ = "14" GOTO 4160
3770 IF L$ = "15" GOTO 4170
3780 IF L$ = "16" GOTO 4180
3790 IF L$ = "A" OR L$ = "a" THEN RETURN
3800 IF L$ <> "N" AND L$ <> "A" AND L$ < "0" OR L$ > "16" THEN GOTO 3360
3810 INPUT " 0 GLAND DASH NUMBER (AS568 OR ISO O-RING CODE) "; SIZE$: GOTO 3360
3820 OLDTYPE$ = TYPE$
3830 INPUT " 1 TYPE: (R)OD-ID SEAL OR (P)ISTON-OD SEAL "; TYPE$
3840 IF TYPE$ = "r" THEN TYPE$ = "R"
3850 IF TYPE$ = "p" THEN TYPE$ = "P"
3860 IF OLDTYPE$ = "R" AND TYPE$ = "P" GOTO 3980
3870 IF OLDTYPE$ = TYPE$ THEN GOTO 3360
3880 REM=====
3890 REM CHANGE GLAND VALUES
3900 REM PISTON TO ROD
3910 REM=====
3920 DRL = DGL: DRH = DGH: DGL = DBL: DGH = DBH: DBL = 0: DBH = 0
3930 GOTO 3360
3940 REM=====
3950 REM CHANGE GLAND VALUES
3960 REM ROD TO PISTON
3970 REM=====
3980 DBL = DGL: DBH = DGH: DGL = DRL: DGH = DRH: DPL = 0: DPH = 0
3990 GOTO 3360
4000 INPUT " 2 SEAL CROSS SECTION - LOW VALUE...."; WL: GOTO 3360
4010 INPUT " 3 SEAL CROSS SECTION - HIGH VALUE...."; WH: GOTO 3360
4020 INPUT " 4 SEAL ID - LOW VALUE....."; IDSL: GOTO 3360
4030 INPUT " 5 SEAL ID - HIGH VALUE....."; IDSH: GOTO 3360
4040 IF TYPE$ = "R" GOTO 4080
4050 INPUT " 6 PISTON DIAMETER - LOW VALUE....."; DPL: GOTO 3360
4060 IF TYPE$ = "R" GOTO 4090
4070 INPUT " 7 PISTON DIAMETER - HIGH VALUE....."; DPH: GOTO 3360
4080 INPUT " 6 ROD DIAMETER - LOW VALUE....."; DRL: GOTO 3360
4090 INPUT " 7 ROD DIAMETER - HIGH VALUE....."; DRH: GOTO 3360
4100 INPUT " 8 BORE DIAMETER - LOW VALUE....."; DBL: GOTO 3360
4110 INPUT " 9 BORE DIAMETER - HIGH VALUE....."; DBH: GOTO 3360
4120 INPUT " 10 GROOVE DIAMETER - LOW VALUE....."; DGL: GOTO 3360
4130 INPUT " 11 GROOVE DIAMETER - HIGH VALUE....."; DGH: GOTO 3360
4140 INPUT " 12 GROOVE ECCENTRICITY "; ECC: GOTO 3360
4150 INPUT " 13 CAP SEAL THICKNESS - LOW VALUE...."; TCAPL: GOTO 3360
4160 INPUT " 14 CAP SEAL THICKNESS - HIGH VALUE..."; TCAPH: GOTO 3360
4170 INPUT " 15 BACKUP RADIAL HEIGHT - HIGH VALUE."; WBUH: GOTO 3360
4180 INPUT " 16 COMMENT: "; COMMENT$: GOTO 3360
4190 REM=====
4200 REM PRINT OUT VARIABLES
4210 REM=====
4220 CLS : LOCATE 1
4230 PRINT TAB(20); "INPUT VARIABLES"
4240 PRINT "SIZE$ = GLAND DASH NUMBER PER AS568 OR ISO O-RING CODE"
4250 PRINT "TYPE$ = (R)OD-ID SEAL OR (P)ISTON-OD SEAL TYPE"
4260 PRINT "WL = SEAL CROSS SECTION - LOW VALUE"
4270 PRINT "WH = SEAL CROSS SECTION - HIGH VALUE"
4280 PRINT "IDSL = SEAL ID - LOW VALUE"
4290 PRINT "IDSH = SEAL ID - HIGH VALUE"
4300 PRINT "DRL = ROD DIAMETER - LOW VALUE"
4310 PRINT "DRH = ROD DIAMETER - HIGH VALUE"
4320 PRINT "DPL = PISTON DIAMETER - LOW VALUE"
4330 PRINT "DPH = PISTON DIAMETER - HIGH VALUE"
4340 PRINT "DBL = BORE DIAMETER - LOW VALUE"

```

SAE ARP4727 Revision A

```

4350 PRINT "DBH = BORE DIAMETER - HIGH VALUE"
4360 PRINT "DGL = GROOVE DIAMETER - LOW VALUE"
4370 PRINT "DGH = GROOVE DIAMETER - HIGH VALUE"
4380 PRINT "ECC = GROOVE ECCENTRICITY = TOTAL INDICATOR READING"
4390 PRINT "TCAPL = CAP SEAL THICKNESS - LOW VALUE"
4400 PRINT "TCAPH = CAP SEAL THICKNESS - HIGH VALUE"
4410 PRINT "WBUH = BACKUP RING RADIAL HEIGHT - HIGH VALUE"
4420 LOCATE 23: PRINT "PRESS (C) TO FINISH LIST OR (Q) TO RETURN TO PROGRAM"; : INPUT A$
4430 IF A$ = "Q" OR A$ = "q" THEN GOTO 630
4440 CLS : LOCATE 1
4450 PRINT TAB(20); "OTHER VARIABLES"
4460 PRINT "SQEL = SQUEEZE IN PERCENT - LOW VALUE"
4470 PRINT "SQEH = SQUEEZE IN PERCENT - HIGH VALUE"
4480 PRINT "STRL = LOW NUMERICAL STRETCH USED FOR MAX SQUEEZE" 'REV A
4490 PRINT "STRH = HIGH NUMERICAL STRETCH USED FOR MIN SQUEEZE" 'REV A
4500 PRINT "GLNDMX = GLAND CROSS SECTION AT MINIMUM SQUEEZE WITH ROD OR"
4510 PRINT "        PISTON CENTERED"
4520 PRINT "GLNDMN = GLAND CROSS SECTION AT MAXIMUM SQUEEZE WITH ROD OR"
4530 PRINT "        PISTON CENTERED"
4540 PRINT "DCLRMX = DIAMETRAL CLEARANCE AT MINIMUM SQUEEZE"
4550 PRINT "DCLRMN = DIAMETRAL CLEARANCE AT MAXIMUM SQUEEZE"
4560 PRINT "RCLRMX = RADIAL CLEARANCE AT MINIMUM SQUEEZE"
4570 PRINT "RCLRMN = RADIAL CLEARANCE AT MAXIMUM SQUEEZE"
4580 PRINT "RECC = ECCENTRICITY / 2"
4590 PRINT "DELWL = FACTOR FOR CHANGE IN MINIMUM SEAL CROSS SECTION"
4600 PRINT "DELWH = FACTOR FOR CHANGE IN MAXIMUM SEAL CROSS SECTION"
4610 PRINT "DEFMN = CROSS-SECTION DEFLECTION AT MINIMUM SQUEEZE"
4620 PRINT "DEFMX = CROSS-SECTION DEFLECTION AT MAXIMUM SQUEEZE"
4630 PRINT "WBUH = BACKUP RING HEIGHT - HIGH VALUE"
4640 PRINT "WMIN = MINIMUM INSTALLED O-RING CROSS SECTION" 'REV A
4650 PRINT "WMAX = MAXIMUM INSTALLED O-RING CROSS SECTION" 'REV A
4660 PRINT "DEFLOI = MINIMUM ALLOWABLE DEFLECTION, INCHES"
4670 PRINT "DEFLOM = MINIMUM ALLOWABLE DEFLECTION, MM"
4680 PRINT "SQUZHI = MAXIMUM ALLOWABLE SQUEEZE - PERCENT"
4690 PRINT "STRCHI = MAXIMUM ALLOWABLE STRETCH - PERCENT"
4700 PRINT "ISOWL = MINIMUM ISO O-RING CROSS-SECTION ROUNDED DOWN TO NEAREST 10TH"
4710 LOCATE 25: PRINT "PRESS (R) TO REPEAT LIST OR (Q) TO RETURN TO PROGRAM"; : INPUT A$
4720 IF A$ = "R" OR A$ = "r" GOTO 4200
4730 GOTO 630
4740 REM=====
4750 REM  CHANGE DEFAULT ADVISORY VALUES
4760 REM=====
4770 REM
4780 CLS
4790 LOCATE 10
4800 PRINT TAB(5); "1  MINIMUM DEFLECTION ADVISORY VALUE-INCHES"; TAB(60); : PRINT USING
"####.####"; DEFLOI
4810 PRINT TAB(5); "2  MINIMUM DEFLECTION ADVISORY VALUE-MM"; TAB(60); : PRINT USING "####.####";
DEFLOM
4820 PRINT TAB(5); "3  MAXIMUM PERCENT SQUEEZE ADVISORY VALUE"; TAB(60); : PRINT USING "####.#";
SQUZHI; : PRINT " %"
4830 PRINT TAB(5); "4  MAXIMUM PERCENT INSTALLED STRETCH ADVISORY VALUE"; TAB(60); : PRINT USING
"####.#"; STRCHI; : PRINT " %"
4840 LOCATE 20: PRINT "        DO YOU WANT TO CHANGE SOME VALUES?"
4850 INPUT "ENTER 'N' FOR NONE OR 1 TO 4 FOR ITEM"; L$
4860 IF L$ = "N" OR L$ = "n" THEN GOTO 630
4870 IF L$ = "1" GOTO 4920
4880 IF L$ = "2" GOTO 4930
4890 IF L$ = "3" GOTO 4940
4900 IF L$ = "4" GOTO 4950
4910 IF L$ < "1" OR L$ > "4" THEN GOTO 4780
4920 INPUT "  1 MINIMUM DEFLECTION ADVISORY VALUE-INCHES  "; DEFLOI: DEFLOM = DEFLOI * 25.4: GOTO
4770 'REV A
4930 INPUT "  2 MINIMUM DEFLECTION ADVISORY VALUE-MM      "; DEFLOM: DEFLOI = DEFLOM / 25.4: GOTO
4770 'REV A
4940 INPUT "  3 MAXIMUM PERCENT SQUEEZE ADVISORY VALUE      "; SQUZHI: GOTO 4770
4950 INPUT "  4 MAXIMUM INSTALLED STRETCH ADVISORY VALUE      "; STRCHI: GOTO 4770

```

SAE ARP4727 Revision A

APPENDIX B LISTING OF "GV.BAS"

Line revisions are denoted by 'Rev A within program

```

10 REM * * * * *
20 REM *      PROGRAM GV.BAS - GLAND VOLUME CALCULATION
30 REM *      VERSION 3.0
40 REM *
50 REM * THIS IS A PROGRAM TO CALCULATE THE MINIMUM AND MAXIMUM VOLUME OF
60 REM * SEAL GLANDS. THE PROGRAM ALSO CALCULATES O-RING AND BACKUP
70 REM * VOLUME TO COMPUTE PERCENT OCCUPANCY.
80 REM *
90 REM *      VERSION 3.0 HAS THE FOLLOWING CAPABILITIES:
100 REM* (A) ALGORITHM IS BASED UPON COMPUTATION OF VOLUMES OF REVOLUTION
110 REM* USING INTEGRAL CALCULUS.
120 REM* (B) COMPUTATION INCLUDES INPUT FOR WALL ANGLE. A NEGATIVE
130 REM* WALL ANGLE IS ALLOWED.
140 REM* (C) BACKUP RING AND O-RING DIMENSIONS ARE INPUT VALUES.
150 REM* (D) MAXIMUM AND MINIMUM GLAND VOLUME AND MAXIMUM AND MINIMUM
160 REM* PERCENT OCCUPANCY ARE OUTPUT.
170 REM* (E) PROGRAM ALLOWS CHOICE OF EITHER PISTON (OD SEAL) OR
180 REM* ROD (ID SEAL) GLAND VOLUME COMPUTATION.
190 REM* (F) SAE OR ISO DIMENSIONS ARE SUPPORTED.
200 REM* (G) VOLUME OF A SPECIAL CONFIGURATION MAY BE ENTERED TO SUPPORT
210 REM* CAP STRIP OR NON-STANDARD SEALS.
220 REM* (H) HARD COPY OF OUTPUT MAY BE OBTAINED BY METHODS 1 OR 2 BELOW.
230 REM* (1) BEFORE RUNNING GV.BAS, CREATE A FOLDER CALLED SEALPRNT
240 REM* ON DRIVE C:\. THE PROGRAM WILL PRINT TO A FILE CALLED
250 REM* GVOUT.TXT WHEN THE OPTION "P" IS PROMPTED.
260 REM* (2) A HARD COPY MAY ALSO BE OBTAINED BY PRESSING THE
270 REM* PRINT SCR N KEY ON THE COMPUTER KEYBOARD WHILE THE SCREEN
280 REM* IS DISPLAYED. THEN OPEN A NEW FILE IN WORD PROCESSOR
290 REM* SOFTWARE, PASTE THE OUTPUT TO THE WORD PROCESSOR FILE.
300 REM* EITHER METHOD ABOVE WILL GIVE AN EDITABLE TEXT FILE WHICH
310 REM* CAN BE PRINTED FROM THE WORD PROCESSOR FILE OR CUT AND
320 REM* PASTED INTO ANOTHER FILE IF DESIRED.
330 REM*
340 REM* ORIGINALLY WRITTEN FEBRUARY-MARCH 1993 BY R.B. OLSEN
350 REM* FOR ARP4727 "COMPUTATION OF SEAL GLAND SQUEEZE AND VOLUME"
360 REM* SAE PROJECT A6C2-91-2
370 REM* REVISED: JANUARY 2003 SAE PROJECT A6C2-01-03
380 REM* EDITED LINE 4420; ADDED LINE 4430; EDITED LINE 4440 TO GIVE
390 REM* CORRECT LOGIC WHEN CASE IS CHANGED FROM P TO R OR R TO P
400 REM*
410 REM* * * * * *
420 REM=====
430 REM  CLEAR SCREEN
440 REM=====
450 CLS
460 LOCATE 2: PRINT TAB(5); "GLAND VOLUME AND PERCENT OCCUPANCY - USING PGM GV.BAS"; TAB(60);
"VERSION 3.0" 'REV A
470 LOCATE 4: PRINT TAB(10); "FOR PROGRAM DESCRIPTION SEE ARP4727 'GLAND DESIGN,"
480 PRINT TAB(10); "COMPUTATION OF SEAL SQUEEZE AND GLAND VOLUME"
490 LOCATE 6: PRINT TAB(10); "AVAILABLE FROM SOCIETY OF AUTOMOTIVE ENGINEERS, WARRENDALE, PA"
500 LOCATE 9: PRINT TAB(5); "NOTE: PROGRAM ALLOWS ENTRY OF MIN AND MAX VOLUME OF SPECIAL SEAL."
510 LOCATE 10: PRINT TAB(5); "BE SURE TO USE CONSISTENT UNITS OF MM FOR ISO OR INCHES OTHERWISE."
520 LOCATE 14: PRINT TAB(10); "(R) RUN PROGRAM; (V) VIEW VARIABLE LIST OR (Q) QUIT"; : INPUT A$
530 IF A$ = "V" OR A$ = "v" THEN GOSUB 4790
540 IF A$ = "Q" OR A$ = "q" THEN END
550 REM=====
560 REM  INITIALIZE CONSTANTS
570 REM=====
580 PI = 3.141592654#
590 RDDEG = PI / 180
600 REM=====
610 REM  INPUT VARIABLES
620 REM=====
630 CLS
640 INPUT " 0 AS568 DASH NUMBER OR ISO O-RING SIZE CODE..... -"; SIZE$

```

SAE ARP4727 Revision A

```

650 INPUT " 1 (R)OD-ID SEAL OR (P)ISTON-OD SEAL..... "; TYPE$
660 IF TYPE$ = "r" THEN TYPE$ = "R"
670 IF TYPE$ = "p" THEN TYPE$ = "P"
680 INPUT " 2 GLAND WIDTH - LOW..... "; WGL
690 INPUT " 3 GLAND WIDTH - HIGH..... "; WGH
700 INPUT " 4 GROOVE DIAMETER - LOW..... "; DGL
710 INPUT " 5 GROOVE DIAMETER - HIGH..... "; DGH
720 INPUT " 6 GROOVE WALL ANGLE, DEGREES - LOW..... "; AGL
730 INPUT " 7 GROOVE WALL ANGLE, DEGREES - HIGH..... "; AGH
740 INPUT " 8 GROOVE CORNER RADIUS - LOW..... "; RGL
750 INPUT " 9 GROOVE CORNER RADIUS - HIGH..... "; RGH
760 IF TYPE$ = "R" GOTO 800
770 INPUT " 10 PISTON BORE DIAMETER - LOW..... "; DBL
780 INPUT " 11 PISTON BORE DIAMETER - HIGH..... "; DBH
790 GOTO 820
800 INPUT " 10 ROD DIAMETER - LOW..... "; DRL
810 INPUT " 11 ROD DIAMETER - HIGH..... "; DRH
820 INPUT " 12 O-RING CROSS SECTION DIAMETER - LOW..... "; WSL
830 INPUT " 13 O-RING CROSS SECTION DIAMETER - HIGH..... "; WSH
840 INPUT " 14 O-RING INNER DIAMETER - LOW..... "; IDSL
850 INPUT " 15 O-RING INNER DIAMETER - HIGH..... "; IDSH
860 INPUT " 16 NUMBER OF BACK UP RINGS..... "; NB
870 IF NB = 0 THEN GOTO 980
880 INPUT " 17 BACK UP RING INNER DIAMETER - LOW..... "; IDBUL
890 INPUT " 18 BACK UP RING INNER DIAMETER - HIGH..... "; IDBUH
900 INPUT " 19 BACK UP RING RADIAL WIDTH - LOW..... "; WBL
910 INPUT " 20 BACK UP RING RADIAL WIDTH - HIGH..... "; WBH
920 INPUT " 21 BACK UP RING THICKNESS - LOW..... "; TBL
930 INPUT " 22 BACK UP RING THICKNESS - HIGH..... "; TBH
940 INPUT " 23 BACK UP RING GAP - LOW..... "; GBUL
950 INPUT " 24 BACK UP RING GAP - HIGH..... "; GBUH
960 INPUT " 25 BACK UP RING GAP ANGLE, DEGREES - LOW..... "; ABUL
970 INPUT " 26 BACK UP RING GAP ANGLE, DEGREES - HIGH..... "; ABUH 'REV A
980 INPUT " 27 MIN VOLUME OF SPECIAL SEAL (ENTER 0 IF NONE).... "; VSPCL
990 IF VSPCL = 0 THEN VSPCH = 0: GOTO 1010
1000 INPUT " 28 MAX VOLUME OF SPECIAL SEAL (ENTER 0 IF NONE).... "; VSPCH
1010 INPUT " 29 COMMENT: "; COMMENT$
1020 REM=====
1030 REM CALL SUBR TO CHECK INPUT VALUES
1040 REM=====
1050 GOSUB 3590
1060 REM=====
1070 REM CONVERT ANGLES IN DEGREES TO RADIANS
1080 REM=====
1090 AGLR = AGL * RDDEG
1100 AGHR = AGH * RDDEG
1110 ABULR = ABUL * RDDEG
1120 ABUHR = ABUH * RDDEG
1130 REM=====
1140 REM COMPUTE VOLUMES 1 AND 5
1150 REM triangle bounded by area efc
1160 REM=====
1170 IF TYPE$ = "R" GOTO 1310
1180 CFL = DBL / 2 - RGH + RGH * SIN(AGLR) - DGH / 2
1190 CFH = DBH / 2 - RGL + RGL * SIN(AGHR) - DGL / 2
1200 CL = DGH / 2 + RGH - RGH * SIN(AGLR)
1210 CH = DGL / 2 + RGL - RGL * SIN(AGHR)
1220 BL = CFL
1230 BH = CFH
1240 EFL = CFL * TAN(AGLR)
1250 EFH = CFH * TAN(AGHR)
1260 AL = EFL
1270 AH = EFH
1280 V1MIN = PI * AL * BL * CL + 2 * (PI * AL * BL ^ 2) / 3
1290 V1MAX = PI * AH * BH * CH + 2 * (PI * AH * BH ^ 2) / 3
1300 GOTO 1430
1310 CL = DRL / 2
1320 CH = DRH / 2
1330 CFL = DGL / 2 - RGH + RGH * SIN(AGLR) - DRH / 2
1340 CFH = DGH / 2 - RGL + RGL * SIN(AGHR) - DRL / 2
1350 BL = CFL

```

SAE ARP4727 Revision A

```

1360 BH = CFH
1370 EFL = CFL * TAN(AGLR)
1380 EFH = CFH * TAN(AGHR)
1390 AL = EFL
1400 AH = EFH
1410 V1MIN = PI * AL * BL * CH + PI * AL * BL ^ 2 / 3
1420 V1MAX = PI * AH * BH * CL + PI * AH * BH ^ 2 / 3
1430 REM=====
1440 REM      COMPUTE VOLUMES 2 AND 6
1450 REM      bounded by area cja
1460 REM=====
1470 IF TYPE$ = "R" GOTO 1590
1480 AL = RGL
1490 AH = RGH
1500 BL = DGL / 2 + RGL
1510 BH = DGH / 2 + RGH
1520 K1 = COS(AGLR)
1530 K2 = COS(AGHR)
1540 IF K1 = 1 THEN K4 = PI / 2 ELSE K4 = ATN(K1 / SQR(-K1 * K1 + 1))
1550 IF K2 = 1 THEN K5 = PI / 2 ELSE K5 = ATN(K2 / SQR(-K2 * K2 + 1))
1560 V2MIN = PI * AH ^ 2 * BH * (K4 - SIN(AGLR) * COS(AGLR)) - 2 * PI * AH ^ 3 * (COS(AGLR)) ^ 3 / 3
1570 V2MAX = PI * AL ^ 2 * BL * (K5 - SIN(AGHR) * COS(AGHR)) - 2 * PI * AL ^ 3 * (COS(AGHR)) ^ 3 / 3
1580 GOTO 1700
1590 AL = RGL
1600 AH = RGH
1610 BL = DGL / 2 - RGH
1620 BH = DGH / 2 - RGL
1630 K1 = COS(AGLR)
1640 K2 = COS(AGHR)
1650 REM      k4 = ARCSIN(k1) and k5 = ARCSIN(k2)
1660 IF K1 = 1 THEN K4 = PI / 2 ELSE K4 = ATN(K1 / SQR(-K1 * K1 + 1))
1670 IF K2 = 1 THEN K5 = PI / 2 ELSE K5 = ATN(K2 / SQR(-K2 * K2 + 1))
1680 V2MIN = PI * AH ^ 2 * BL * (K4 - SIN(AGLR) * COS(AGLR)) + 2 * PI * AH ^ 3 * (COS(AGLR)) ^ 3 / 3
1690 V2MAX = PI * AL ^ 2 * BH * (K5 - SIN(AGHR) * COS(AGHR)) + 2 * PI * AL ^ 3 * (COS(AGHR)) ^ 3 / 3
1700 REM=====
1710 REM      COMPUTE VOLUMES 4 AND 7
1720 REM      bounded by area fgjc
1730 REM=====
1740 IF TYPE$ = "R" GOTO 1860
1750 AL = RGL * COS(AGHR)
1760 AH = RGH * COS(AGLR)
1770 CFL = DBL / 2 - RGH + RGL * SIN(AGLR) - DGH / 2
1780 CFH = DBH / 2 - RGL + RGL * SIN(AGHR) - DGL / 2
1790 BL = CFL
1800 BH = CFH
1810 CL = DGL / 2 + RGL - RGL * SIN(AGHR)
1820 CH = DGH / 2 + RGH - RGH * SIN(AGLR)
1830 V4MIN = 2 * PI * AH * BL * CH + PI * AH * BL ^ 2
1840 V4MAX = 2 * PI * AL * BH * CL + PI * AL * BH ^ 2
1850 GOTO 1960
1860 AL = RGL * COS(AGHR)
1870 AH = RGH * COS(AGLR)
1880 CFL = DGL / 2 - RGH + RGH * SIN(AGLR) - DRH / 2
1890 CFH = DGH / 2 - RGL + RGL * SIN(AGHR) - DRL / 2
1900 BL = CFL
1910 BH = CFH
1920 CL = DRL / 2
1930 CH = DRH / 2
1940 V4MIN = 2 * PI * AH * BL * CH + PI * AH * BL ^ 2
1950 V4MAX = 2 * PI * AL * BH * CL + PI * AL * BH ^ 2
1960 REM=====
1970 REM      COMPUTE VOLUME 3
1980 REM      bounded by area ihag
1990 REM=====
2000 IF TYPE$ = "R" GOTO 2120
2010 ABL = RGL * TAN((PI - 2 * AGHR) / 4)
2020 ABH = RGH * TAN((PI - 2 * AGLR) / 4)

```

SAE ARP4727 Revision A

```

2030 AL = WGL - 2 * ABH
2040 AH = WGH - 2 * ABL
2050 BL = (DBL - DGH) / 2
2060 BH = (DBH - DGL) / 2
2070 CL = DGL / 2
2080 CH = DGH / 2
2090 V3MIN = 2 * PI * AL * BL * CH + PI * AL * BL ^ 2
2100 V3MAX = 2 * PI * AH * BH * CL + PI * AH * BH ^ 2
2110 GOTO 2220
2120 ABL = RGL * TAN((PI - 2 * AGHR) / 4)
2130 ABH = RGH * TAN((PI - 2 * AGLR) / 4)
2140 AL = WGL - 2 * ABH
2150 AH = WGH - 2 * ABL
2160 BL = (DGL - DRH) / 2
2170 BH = (DGH - DRL) / 2
2180 CL = DRL / 2
2190 CH = DRH / 2
2200 V3MIN = 2 * PI * AL * BL * CH + PI * AL * BL ^ 2
2210 V3MAX = 2 * PI * AH * BH * CL + PI * AH * BH ^ 2
2220 REM=====
2230 REM COMPUTE VOLUME OF O-RING
2240 REM=====
2250 VSMIN = (PI ^ 2 * (IDSL + WSL) * WSL ^ 2) / 4
2260 VSMAX = (PI ^ 2 * (IDSH + WSH) * WSH ^ 2) / 4
2270 REM=====
2280 REM COMPUTE VOLUME OF BACKUP RING
2290 REM=====
2300 IF ABULR = 0 THEN IDBL = IDBUL ELSE IDBL = IDBUL - GBUH / (PI * SIN(ABULR))
2310 IF ABULR = 0 THEN IDBH = IDBUH ELSE IDBH = IDBUH - GBUL / (PI * SIN(ABUHR))
2320 VBMIN = PI * (WBL * IDBL + WBL ^ 2) * TBL
2330 VBMAX = PI * (WBH * IDBH + WBH ^ 2) * TBH
2340 REM=====
2350 REM COMPUTE VOLUME OF GLAND
2360 REM=====
2370 VGLDMN = (V1MIN + V2MIN + V4MIN) * 2 + V3MIN
2380 VGLDMX = (V1MAX + V2MAX + V4MAX) * 2 + V3MAX
2390 REM=====
2400 REM COMPUTE PERCENT OCCUPANCY
2410 REM=====
2420 PCTMIN = 100 * (VBMIN * NB + VSMIN * VSPCL) / VGLDMX
2430 PCTMAX = 100 * (VBMAX * NB + VSMAX * VSPCH) / VGLDMN
2440 REM=====
2450 REM PRINT RESULTS TO SCREEN
2460 REM=====
2470 IF WSL > 1 THEN DM1$ = "-mm" ELSE DM1$ = "-inches"
2480 IF WSL > 1 THEN DM2$ = "-mm^3" ELSE DM2$ = "-in^3"
2490 CLS : LOCATE 1
2500 TYPE1$ = TYPE$ + "OD (ID SEAL)": IF TYPE$ = "P" THEN TYPE1$ = TYPE$ + "ISTON (OD SEAL)"
2510 PRINT TAB(8); "GLAND VOLUME AND PERCENT OCCUPANCY - USING PGM GV.BAS VERSION 1.0"
2520 PRINT TAB(2); "GLAND IDENTIFICATION (AS568 OR ISO O-RING CODE) -"; SIZE$; TAB(58); "TYPE: ";
TYPE1$
2530 PRINT TAB(2); "COMMENT: "; COMMENT$
2540 PRINT TAB(42); " LOW"; TAB(62); " HIGH"
2550 PRINT TAB(40); "-----"; TAB(60); "-----"
2560 PRINT TAB(5); "GROOVE WIDTH"; DM1$; TAB(40); : PRINT USING "####.####"; WGL; : PRINT
TAB(60); : PRINT USING "####.####"; WGH
2570 PRINT TAB(5); "GROOVE DIA"; DM1$; TAB(40); : PRINT USING "####.####"; DGL; : PRINT TAB(60);
: PRINT USING "####.####"; DGH
2580 PRINT TAB(5); "GROOVE WALL ANGLE-DEG"; TAB(40); : PRINT USING "####.####"; AGL; : PRINT
TAB(60); : PRINT USING "####.####"; AGH
2590 PRINT TAB(5); "GROOVE CORNER RADIUS"; DM1$; TAB(40); : PRINT USING "####.####"; RGL; : PRINT
TAB(60); : PRINT USING "####.####"; RGH
2600 IF TYPE$ = "P" GOTO 2630
2610 PRINT TAB(5); "ROD DIAMETER"; DM1$; TAB(40); : PRINT USING "####.####"; DRL; : PRINT
TAB(60); : PRINT USING "####.####"; DRH
2620 GOTO 2640
2630 PRINT TAB(5); "BORE DIAMETER"; DM1$; TAB(40); : PRINT USING "####.####"; DBL; : PRINT
TAB(60); : PRINT USING "####.####"; DBH
2640 PRINT TAB(5); "O-RING CROSS-SECTION DIA."; DM1$; TAB(40); : PRINT USING "####.####"; WSL; :
PRINT TAB(60); : PRINT USING "####.####"; WSH

```

SAE ARP4727 Revision A

```

2650 PRINT TAB(5); "O-RING INNER DIAMETER"; DM1$; TAB(40); : PRINT USING "####.####"; IDSL; :
PRINT TAB(60); : PRINT USING "####.####"; IDSH
2660 PRINT TAB(5); "NUMBER OF BACKUP RINGS"; TAB(42); : PRINT USING "####"; NB
2670 IF NB = 0 THEN GOTO 2750
2680 PRINT TAB(5); "BACKUP RING INNER DIAMETER"; DM1$; TAB(40); : PRINT USING "####.####"; IDBUL;
: PRINT TAB(60); : PRINT USING "####.####"; IDBUH
2690 PRINT TAB(5); "BACKUP RING RADIAL WIDTH"; DM1$; TAB(40); : PRINT USING "####.####"; WBL; :
PRINT TAB(60); : PRINT USING "####.####"; WBH
2700 PRINT TAB(5); "BACKUP RING THICKNESS"; DM1$; TAB(40); : PRINT USING "####.####"; TBL; :
PRINT TAB(60); : PRINT USING "####.####"; TBH
2710 IF GBUL = 0 AND GBUH = 0 THEN GOTO 2750
2720 PRINT TAB(5); "BACKUP RING GAP"; DM1$; TAB(40); : PRINT USING "####.####"; GBUL; : PRINT
TAB(60); : PRINT USING "####.####"; GBUH
2730 PRINT TAB(5); "BACKUP RING GAP ANGLE-DEGREES"; TAB(40); : PRINT USING "####.####"; ABUL; :
PRINT TAB(60); : PRINT USING "####.####"; ABUH
2740 LOCATE 23
2750 INPUT "ENTER <CR> TO CONTINUE SCREEN OUTPUT"; A$
2760 CLS : LOCATE 2
2770 PRINT TAB(40); "VOLUMES AND PERCENT OCCUPANCY"
2780 PRINT TAB(40); "  MIN PCT"; TAB(60); "  MAX PCT"
2790 PRINT TAB(40); "-----"; TAB(60); "-----"
2800 PRINT TAB(15); "V1"; DM2$; TAB(40); : PRINT USING "#.#####^"; V1MAX; : PRINT TAB(60); :
PRINT USING "#.#####^"; V1MIN
2810 PRINT TAB(15); "V2"; DM2$; TAB(40); : PRINT USING "#.#####^"; V2MAX; : PRINT TAB(60); :
PRINT USING "#.#####^"; V2MIN
2820 PRINT TAB(15); "V3"; DM2$; TAB(40); : PRINT USING "#.#####^"; V3MAX; : PRINT TAB(60); :
PRINT USING "#.#####^"; V3MIN
2830 PRINT TAB(15); "V4"; DM2$; TAB(40); : PRINT USING "#.#####^"; V4MAX; : PRINT TAB(60); :
PRINT USING "#.#####^"; V4MIN
2840 PRINT TAB(15); "GLAND VOLUME"; DM2$; TAB(40); : PRINT USING "#.#####^"; VGLDMX; : PRINT
TAB(60); : PRINT USING "#.#####^"; VGLDMN
2850 PRINT TAB(15); "SEAL VOLUME"; DM2$; TAB(40); : PRINT USING "#.#####^"; VSMIN; : PRINT
TAB(60); : PRINT USING "#.#####^"; VSMAX
2860 PRINT TAB(15); "SPECIAL SEAL VOLUME"; DM2$; TAB(40); : PRINT USING "#.#####^"; VSPCL; :
PRINT TAB(60); : PRINT USING "#.#####^"; VSPCH
2870 PRINT TAB(15); "BACKUP VOLUME"; DM2$; TAB(40); : PRINT USING "#.#####^"; VBMIN; : PRINT
TAB(60); : PRINT USING "#.#####^"; VBMAX
2880 PRINT TAB(15); "PCT OCCUPANCY"; TAB(40); : PRINT USING "###.#####"; PCTMIN; : PRINT
TAB(60); : PRINT USING "###.#####"; PCTMAX
2890 PRINT TAB(15); "***** ANALYSIS COMPLETE *****"
2900 REM=====
2910 REM      OPTIONS MENU
2920 REM=====
2930 LOCATE 22
2940 PRINT TAB(20); : PRINT "CHOOSE OPTION DESIRED BELOW":
2950 PRINT TAB(10); : INPUT "(P)RINT (Q)UIT (R)UN AGAIN "; B$
2960 IF B$ = "P" OR B$ = "p" THEN 3000
2970 IF B$ = "Q" OR B$ = "q" THEN 3560
2980 IF B$ = "R" OR B$ = "r" THEN GOTO 3580
2990 IF ((B$ <> "P") AND (B$ <> "Q") AND (B$ <> "p") AND (B$ <> "q")) THEN 2930
3000 REM=====
3010 REM      SEND RESULTS TO PRINTER
3020 REM=====
3030 OPEN "C:\SEALPRNT\GVOUT.TXT" FOR OUTPUT AS #1 'REV A
3040 T$ = TIME$ 'REV A
3050 D$ = DATE$ 'REV A
3060 HR = VAL(T$) 'REV A
3070 IF HR < 12 THEN AMPM$ = " AM" ELSE AMPM$ = " PM" 'REV A
3080 IF HR > 12 THEN HR = HR - 12 'REV A
3090 PRINT #1, TAB(8); "GLAND VOLUME AND PERCENT OCCUPANCY - USING PGM GV.BAS" 'REV A
3100 PRINT #1, 'REV A
3110 PRINT #1, TAB(8); "Time: "; STR$(HR); RIGHT$(T$, 6); AMPM$; TAB(28); "Date: "; D$ 'REV A
3120 PRINT #1, 'REV A
3130 PRINT #1, TAB(2); "GLAND IDENTIFICATION (AS568 OR ISO O-RING CODE) -"; SIZE$; TAB(58);
"TYPE: "; TYPE1$ 'REV A
3140 PRINT #1, 'REV A
3150 PRINT #1, TAB(2); "COMMENT: "; COMMENT$ 'REV A
3160 PRINT #1, 'REV A
3170 PRINT #1, : FOR I = 0 TO 79: PRINT #1, " "; : NEXT I 'REV A
3180 PRINT #1, 'REV A
3190 PRINT #1, TAB(40); "  LOW"; TAB(60); "  HIGH" 'REV A

```

SAE ARP4727 Revision A

```

3200 PRINT #1, TAB(40); "-----"; TAB(60); "-----" 'REV A
3210 PRINT #1, TAB(5); "GROOVE WIDTH"; DM1$; TAB(40); : PRINT #1, USING "#####"; WGL; : PRINT
#1, TAB(60); : PRINT #1, USING "#####"; WGH 'REV A
3220 PRINT #1, TAB(5); "GROOVE DIA"; DM1$; TAB(40); : PRINT #1, USING "#####"; DGL; : PRINT
#1, TAB(60); : PRINT #1, USING "#####"; DGH 'REV A
3230 PRINT #1, TAB(5); "GROOVE WALL ANGLE-DEG"; TAB(40); : PRINT #1, USING "#####"; AGL; :
PRINT #1, TAB(60); : PRINT #1, USING "#####"; AGH 'REV A
3240 PRINT #1, TAB(5); "GROOVE CORNER RADIUS"; DM1$; TAB(40); : PRINT #1, USING "#####"; RGL;
: PRINT #1, TAB(60); : PRINT #1, USING "#####"; RGH 'REV A
3250 IF TYPE$ = "P" GOTO 3280
3260 PRINT #1, TAB(5); "ROD DIAMETER"; DM1$; TAB(40); : PRINT #1, USING "#####"; DRL; : PRINT
#1, TAB(60); : PRINT #1, USING "#####"; DRH 'REV A
3270 GOTO 3290
3280 PRINT #1, TAB(5); "BORE DIAMETER"; DM1$; TAB(40); : PRINT #1, USING "#####"; DBL; :
PRINT #1, TAB(60); : PRINT #1, USING "#####"; DBH 'REV A
3290 PRINT #1, TAB(5); "O-RING CROSS-SECTION DIA."; DM1$; TAB(40); : PRINT #1, USING "#####";
WSL; : PRINT #1, TAB(60); : PRINT #1, USING "#####"; WSH 'REV A
3300 PRINT #1, TAB(5); "O-RING INNER DIAMETER"; DM1$; TAB(40); : PRINT #1, USING "#####";
IDSL; : PRINT #1, TAB(60); : PRINT #1, USING "#####"; IDSH 'REV A
3310 IF NB = 0 THEN GOTO 3380
3320 PRINT #1, TAB(5); "BACKUP RING INNER DIAMETER"; DM1$; TAB(40); : PRINT #1, USING
"#####"; IDBUL; : PRINT #1, TAB(60); : PRINT #1, USING "#####"; IDBUH 'REV A
3330 PRINT #1, TAB(5); "BACKUP RING WIDTH"; DM1$; TAB(40); : PRINT #1, USING "#####"; WBL; :
PRINT #1, TAB(60); : PRINT #1, USING "#####"; WBH 'REV A
3340 PRINT #1, TAB(5); "BACKUP RING THICKNESS"; DM1$; TAB(40); : PRINT #1, USING "#####";
TBL; : PRINT #1, TAB(60); : PRINT #1, USING "#####"; TBH 'REV A
3350 IF GBUL = 0 AND GBUH = 0 THEN GOTO 3380
3360 PRINT #1, TAB(5); "BACKUP RING GAP"; DM1$; TAB(40); : PRINT #1, USING "#####"; GBUL; :
PRINT #1, TAB(60); : PRINT #1, USING "#####"; GBUH 'REV A
3370 PRINT #1, TAB(5); "BACKUP RING GAP ANGLE-DEG"; TAB(40); : PRINT #1, USING "#####"; ABUL;
: PRINT #1, TAB(60); : PRINT #1, USING "#####"; ABUH 'REV A
3380 PRINT #1, TAB(5); "NUMBER OF BACKUP RINGS"; TAB(42); : PRINT #1, USING "#####"; NB 'REV A
3390 PRINT #1,
3400 PRINT #1, TAB(37); "VOLUMES COMPUTED AND PERCENT OCCUPANCY" 'REV A
3410 PRINT #1, TAB(40); " MIN PCT"; TAB(60); " MAX PCT" 'REV A
3420 PRINT #1, TAB(40); "-----"; TAB(60); "-----" 'REV A
3430 PRINT #1, TAB(15); "V1"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; V1MAX; : PRINT #1,
TAB(60); : PRINT #1, USING "#.#####^"; V1MIN 'REV A
3440 PRINT #1, TAB(15); "V2"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; V2MAX; : PRINT #1,
TAB(60); : PRINT #1, USING "#.#####^"; V2MIN 'REV A
3450 PRINT #1, TAB(15); "V3"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; V3MAX; : PRINT #1,
TAB(60); : PRINT #1, USING "#.#####^"; V3MIN 'REV A
3460 PRINT #1, TAB(15); "V4"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; V4MAX; : PRINT #1,
TAB(60); : PRINT #1, USING "#.#####^"; V4MIN 'REV A
3470 PRINT #1, TAB(15); "GLAND VOLUME"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; VGLDMX;
: PRINT #1, TAB(60); : PRINT #1, USING "#.#####^"; VGLDMN 'REV A
3480 PRINT #1, TAB(15); "SEAL VOLUME"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; VSMIN; :
PRINT #1, TAB(60); : PRINT #1, USING "#.#####^"; VSMAX 'REV A
3490 PRINT #1, TAB(15); "SPECIAL SEAL VOLUME"; DM2$; TAB(40); : PRINT #1, USING "#.#####^";
VSPCL; : PRINT #1, TAB(60); : PRINT #1, USING "#.#####^"; VSPCH 'REV A
3500 PRINT #1, TAB(15); "BACKUP VOLUME"; DM2$; TAB(40); : PRINT #1, USING "#.#####^"; VBMIN;
: PRINT #1, TAB(60); : PRINT #1, USING "#.#####^"; VBMAX 'REV A
3510 PRINT #1, TAB(15); "PCT OCCUPANCY"; TAB(40); : PRINT #1, USING "###.###"; PCTMIN; : PRINT #1,
TAB(60); : PRINT #1, USING "###.###"; PCTMAX 'REV A
3520 PRINT #1, 'REV A
3530 PRINT #1, : FOR I = 0 TO 79: PRINT #1, "*"; : NEXT I
3540 CLOSE 'REV A
3550 GOTO 2760
3560 END
3570 REM=====
3580 REM SCREEN LISTING OF INPUTS
3590 REM AND CORRECTION SUBROUTINE
3600 REM=====
3610 CLS : LOCATE 1
3620 PRINT TAB(20); "PROGRAM GV.BAS - GLAND VOLUME CALCULATION"; : PRINT
3630 PRINT TAB(5); "0 GLAND DASH NUMBER (AS568 OR ISO O-RING SIZE CODE) . . . -"; SIZE$
3640 PRINT TAB(5); "1 TYPE APPLICATION. . . . ."; TYPE$
3650 PRINT TAB(5); "2 GLAND WIDTH - LOW VALUE . . . . ."; : PRINT USING
"#####"; WGL
3660 PRINT TAB(5); "3 GLAND WIDTH - HIGH VALUE. . . . ."; : PRINT USING
"#####"; WGH

```

SAE ARP4727 Revision A

```

3670 PRINT TAB(5); "4  GROOVE DIA - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DGL
3680 PRINT TAB(5); "5  GROOVE DIA - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DGH
3690 PRINT TAB(5); "6  GROOVE WALL ANGLE - LOW VALUE . . . . ."; : PRINT USING
"#####.#####"; AGL
3700 PRINT TAB(5); "7  GROOVE WALL - HIGH VALUE. . . . ."; : PRINT USING
"#####.#####"; AGH
3710 PRINT TAB(5); "8  GROOVE CORNER RADIUS - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; RGL
3720 PRINT TAB(5); "9  GROOVE CORNER RADIUS - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; RGH
3730 IF TYPE$ = "R" GOTO 3770
3740 PRINT TAB(4); "10 PISTON BORE DIAMETER - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DBL
3750 PRINT TAB(4); "11 PISTON BORE DIAMETER - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DBH
3760 GOTO 3790
3770 PRINT TAB(4); "10  ROD DIAMETER - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; DRL
3780 PRINT TAB(4); "11  ROD DIAMETER - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; DRH
3790 PRINT TAB(4); "12  O-RING CROSS SECTION DIA - LOW VALUE. . . . ."; : PRINT USING
"#####.#####"; WSL
3800 PRINT TAB(4); "13  O-RING CROSS SECTION DIA - HIGH VALUE . . . . ."; : PRINT USING
"#####.#####"; WSH
3810 PRINT TAB(4); "14  O-RING INNER DIAMETER - LOW VALUE . . . . ."; : PRINT USING
"#####.#####"; IDSL
3820 PRINT TAB(4); "15  O-RING INNER DIAMETER - HIGH VALUE. . . . ."; : PRINT USING
"#####.#####"; IDSH
3830 LOCATE 20: PRINT "DO YOU WANT TO CHANGE SOME VALUES?"
3840 INPUT "ENTER 'N' FOR NONE. 0 TO 15 FOR ITEM, OR 'A' FOR ALL"; L$
3850 IF L$ = "N" OR L$ = "n" GOTO 4040
3860 REM IF L$ <> "N" AND L$ <> "A" AND L$ < "0" OR L$ > "15" THEN GOTO 3210
3870 IF L$ = "A" OR L$ = "a" THEN GOTO 630
3880 IF L$ = "0" GOTO 4390
3890 IF L$ = "1" GOTO 4400
3900 IF L$ = "2" GOTO 4450
3910 IF L$ = "3" GOTO 4460
3920 IF L$ = "4" GOTO 4470
3930 IF L$ = "5" GOTO 4480
3940 IF L$ = "6" GOTO 4490
3950 IF L$ = "7" GOTO 4500
3960 IF L$ = "8" GOTO 4510
3970 IF L$ = "9" GOTO 4520
3980 IF L$ = "10" GOTO 4530
3990 IF L$ = "11" GOTO 4550
4000 IF L$ = "12" GOTO 4590
4010 IF L$ = "13" GOTO 4600
4020 IF L$ = "14" GOTO 4610
4030 IF L$ = "15" GOTO 4620
4040 CLS : LOCATE 2
4050 PRINT TAB(4); "16  NUMBER OF BACK UP RINGS. . . . ."; NB
4060 IF NB = 0 THEN GOTO 4190
4070 PRINT TAB(4); "17  BACK UP RING INNER DIAMETER - LOW. . . . ."; : PRINT
USING "#####.#####"; IDBUL
4080 PRINT TAB(4); "18  BACK UP RING INNER DIAMETER - HIGH . . . . ."; : PRINT
USING "#####.#####"; IDBUH
4090 PRINT TAB(4); "19  BACK UP RING RADIAL WIDTH - LOW. . . . ."; : PRINT
USING "#####.#####"; WBL
4100 PRINT TAB(4); "20  BACK UP RING RADIAL WIDTH - HIGH . . . . ."; : PRINT
USING "#####.#####"; WBH
4110 PRINT TAB(4); "21  BACK UP RING THICKNESS - LOW . . . . ."; : PRINT
USING "#####.#####"; TBL
4120 PRINT TAB(4); "22  BACK UP RING THICKNESS - HIGH. . . . ."; : PRINT
USING "#####.#####"; TBH
4130 PRINT TAB(4); "23  BACK UP RING  GAP - LOW. . . . ."; : PRINT
USING "#####.#####"; GBUL
4140 PRINT TAB(4); "24  BACK UP RING  GAP - HIGH . . . . ."; : PRINT
USING "#####.#####"; GBUH

```

SAE ARP4727 Revision A

```

4150 PRINT TAB(4); "25 BACK UP RING GAP ANGLE, DEGREES - LOW . . . . ."; : PRINT
USING "####.####"; ABUL
4160 PRINT TAB(4); "26 BACK UP RING GAP ANGLE, DEGREES - HIGH. . . . ."; : PRINT
USING "####.####"; ABUH
4170 PRINT TAB(4); "27 VOLUME OF SPECIAL SEAL - LOW . . . . ."; : PRINT
USING "####.####"; VSPCL
4180 PRINT TAB(4); "28 VOLUME OF SPECIAL SEAL - HIGH. . . . ."; : PRINT
USING "####.####"; VSPCH
4190 PRINT TAB(4); "29 COMMENT: "; COMMENT$
4200 LOCATE 20: PRINT "DO YOU WANT TO CHANGE SOME VALUES?"
4210 INPUT "ENTER 'N' TO EXECUTE, 16 TO 29 FOR ITEM, OR 'A' FOR ALL"; L$
4220 IF L$ = "A" OR L$ = "a" THEN GOTO 630
4230 IF L$ = "N" OR L$ = "n" THEN GOTO 1070
4240 IF L$ = "16" GOTO 4630
4250 IF L$ = "17" GOTO 4650
4260 IF L$ = "18" GOTO 4660
4270 IF L$ = "19" GOTO 4670
4280 IF L$ = "20" GOTO 4680
4290 IF L$ = "21" GOTO 4690
4300 IF L$ = "22" GOTO 4700
4310 IF L$ = "23" GOTO 4710
4320 IF L$ = "24" GOTO 4720
4330 IF L$ = "25" GOTO 4730
4340 IF L$ = "26" GOTO 4740
4350 IF L$ = "27" GOTO 4750
4360 IF L$ = "28" GOTO 4760
4370 IF L$ = "29" GOTO 4770
4380 IF L$ <> "N" AND L$ <> "A" AND L$ < "0" OR L$ > "29" THEN GOTO 4040
4390 INPUT " 0 GLAND DASH NUMBER (AS568 OR ISO O-RING SIZE CODE) .-"; SIZE$: GOTO 4040
4400 OLDTYPE$ = TYPE$
4410 INPUT " 1 TYPE: (R)OD-ID SEAL OR (P)ISTON-OD SEAL....."; TYPE$
4420 IF (OLDTYPE$ = "R" OR OLDTYPE$ = "r") AND (TYPE$ = "P" OR TYPE$ = "p") GOTO 660 'REV A
4430 IF (OLDTYPE$ = "P" OR OLDTYPE$ = "p") AND (TYPE$ = "R" OR TYPE$ = "r") GOTO 660 'REV A
4440 GOTO 3590 'REV A
4450 INPUT " 2 GLAND WIDTH - LOW VALUE....."; WGL: GOTO 3590
4460 INPUT " 3 GLAND WIDTH - HIGH VALUE....."; WGH: GOTO 3590
4470 INPUT " 4 GROOVE DIA - LOW VALUE....."; DGL: GOTO 3590
4480 INPUT " 5 GROOVE DIA - HIGH VALUE....."; DGH: GOTO 3590
4490 INPUT " 6 GROOVE WALL ANGLE, DEG - LOW VALUE....."; AGL: AGLR = AGL * RDDEG:
GOTO 3590
4500 INPUT " 7 GROOVE WALL ANGLE, DEG - HIGH VALUE....."; AGH: AGHR = AGH * RDDEG:
GOTO 3590
4510 INPUT " 8 GROOVE CORNER RADIUS - LOW VALUE....."; RGL: GOTO 3590
4520 INPUT " 9 GROOVE CORNER RADIUS - HIGH VALUE....."; RGH: GOTO 3590
4530 IF TYPE$ = "R" GOTO 4570
4540 INPUT " 10 PISTON BORE DIAMETER - LOW VALUE....."; DBL: GOTO 3590
4550 IF TYPE$ = "R" GOTO 4580
4560 INPUT " 11 PISTON BORE DIAMETER - HIGH VALUE....."; DBH: GOTO 3590
4570 INPUT " 10 ROD DIAMETER - LOW VALUE....."; DRL: GOTO 3590
4580 INPUT " 11 ROD DIAMETER - HIGH VALUE....."; DRH: GOTO 3590
4590 INPUT " 12 O-RING CROSS SECTION DIAMETER - LOW ....."; WSL: GOTO 3590
4600 INPUT " 13 O-RING CROSS SECTION DIAMETER - HIGH ....."; WSH: GOTO 3590
4610 INPUT " 14 O-RING INNER DIAMETER - LOW....."; IDSL: GOTO 3590
4620 INPUT " 15 O-RING INNER DIAMETER - HIGH....."; IDSH: GOTO 3590
4630 INPUT " 16 NUMBER OF BACK UP RINGS....."; NB: GOTO 4040
4640 IF NB = 0 THEN GOTO 1020
4650 INPUT " 17 BACK UP RING INNER DIAMETER - LOW....."; IDBUL: GOTO 4040
4660 INPUT " 18 BACK UP RING INNER DIAMETER - HIGH....."; IDBUH: GOTO 4040
4670 INPUT " 19 BACK UP RING RADIAL WIDTH - LOW....."; WBL: GOTO 4040
4680 INPUT " 20 BACK UP RING RADIAL WIDTH - HIGH....."; WBH: GOTO 4040
4690 INPUT " 21 BACK UP RING THICKNESS - LOW....."; TBL: GOTO 4040
4700 INPUT " 22 BACK UP RING THICKNESS - HIGH....."; TBH: GOTO 4040
4710 INPUT " 23 BACK UP RING GAP - LOW....."; GBUL: GOTO 4040
4720 INPUT " 24 BACK UP RING GAP - HIGH....."; GBUH: GOTO 4040
4730 INPUT " 25 BACK UP RING GAP ANGLE DEGREES - LOW....."; ABUL: ABULR = ABUL *
RDDEG: GOTO 4040
4740 INPUT " 26 BACK UP RING GAP ANGLE DEGREES - HIGH....."; ABUH: ABUHR = ABUH *
RDDEG: GOTO 4040
4750 INPUT " 27 VOLUME OF SPECIAL SEAL - LOW....."; VSPCL: GOTO 4040
4760 INPUT " 28 VOLUME OF SPECIAL SEAL - HIGH....."; VSPCH: GOTO 4040
4770 INPUT " 29 COMMENT: "; COMMENT$: GOTO 4040

```